# 特殊函数差分差商积分及函数正交性与函数范数的Mizar实现研究

## 摘 要

从定理自动证明的发展过程中产生和发展起来的 Mizar 系统，是用来构建Mizar 数学知识库的证明校验系统，也是一个数学定理证明工具。Mizar 系统的逻辑框架基于 Jaśkowski 自然演绎推理的古典逻辑。Mizar 如今已经成为集逻辑证明、校验、排版功能于一体的数学知识处理的形式化系统，拥有自己的数据库 MML，其中收录了一千多篇数学论文，几乎涵盖了数学科学的各个分支。

本文首先介绍了定理机器证明及Mizar系统的历史，其次对如何利用Mizar语言完成数学论文和进行自动推理校验给出了简要的说明。本文主要研究了以Mizar系统为平台，证明特殊函数的差分差商、特殊函数的积分和实现函数正交性与函数范数。本文主要的创新点如下：

1．完成了特殊函数的差分差商在Mizar系统中的实现；

2．根据特殊函数的微分公式及可积性，在Mizar中实现多种特殊函数的积分；

3．首次在Mizar中定义了函数的正交性与函数范数，并讨论了有关性质，应用Mizar语言完成了相关定理的自动推理证明，并验证其正确性。

以上结果均通过Mizar系统的验证，被收录到最新的Mizar数据库（MML）中，并发表于2008年和2009年的***Formalized Mathematics***。

**关键词**：Mizar　定理机器证明　差分差商　积分　函数正交性　函数范数

# THE SPECIAL FUNCTION DIFFERENCE, DIFFERENCE QUOTIENT, INTEGRAL AND ORTHOGONAL, NORM FUNCTION IN MIZAR

## ABSTRACT

The Mizar project is created and develops with the development of automated theorem proving which is a branch of artificial intelligence. Mizar is a proof checker which is used in building Mizar Mathematical Library. The original Mizar is designed to help mathematicians write mathematical papers and it bases on the Jaśkowski style of natural deduction. Today Mizar has become a formalization of mathematical knowledge, with the function of proving, verifying and typesetting. Mizar system has a large Mizar Mathematical Library, including more than one thousand articles, almost covering all branches of mathematics.

This dissertation describes the history of theorem automatic proving and Mizar system and how to write articles in Mizar system. The main work of the paper is summarized as following:

1. Difference and difference quotient of some special functions are implemented in Mizar system.

2. Based on differentials and integrability of special functions, several integrals are presented and implemented in Mizar system.

3. For the first time the orthogonality of functions and the norm of functions are defined. In addition, the properties of related theorems are verified in Mizar system.

The automatic reasoning and proving process related to the above aspects have been completed and verified to be correct in Mizar system.

KEY WORDS: Mizar automated theorem proving difference and difference quotient integral orthogonality norm

# 声　明

## 独创性声明

本人声明所呈交的论文是我个人在导师指导下进行的研究工作及取得的研究成果。尽我所知，除了文中特别加以标注和致谢中所罗列的内容以外，论文中不包含其他人已经发表或撰写过的研究成果，也不包含本人已用于其他学位申请的论文或成果。与我一同工作的同志对本研究所做的任何贡献均已在论文中做了明确的说明并表示了谢意。

申请学位论文与资料若有不实之处，本人承担一切相关责任。

本人签名：　　　　　　　　　　日期：　　　年　　月　　日

## 关于论文使用授权的说明

本学位论文作者完全了解青岛科技大学有关保留、使用学位论文的规定，有权保留并向国家有关部门或机构送交论文的复印件和磁盘，允许论文被查阅和借阅。本人授权学校可以将学位论文的全部或部分内容编入有关数据库进行检索，可以采用影印、缩印或扫描等复制手段保存、汇编学位论文。本人离校后发表或使用学位论文或与该论文直接相关的学术论文或成果时，署名单位仍然为青岛科技大学。(保密的学位论文在解密后适用本授权书)

本学位论文属于：

保密 □，在　　　　年解密后适用于本声明。

不保密 □。

（请在以上方框内打“√”）

本人签名：　　　　　　　　　　日期：　　　年　　月　　日

导师签名：李博　　　　　　　　日期：2009年6月13日

# 1 绪论

## 1.1 定理机器证明(Automated Theorem Proving)的历史及研究现状

数学不但是整个人类文化的重要组成部分，而且始终是推进人类文明的重要力量[1]。数学是对现实世界数与形的最简洁有效的描述，也是高新技术的理论基础，是联系科学与技术的纽带。数学具有表达精确，论证严谨的特点，其主要形式是数值计算和定理证明[2]。

电子计算机是人脑的延伸，它的飞速发展为人类实现脑力劳动的机械化创造了物质条件。实现脑力劳动的机械化，将使科研工作者摆脱繁重的甚至是人力难以胜任的工作，为在科学和高新技术方面进行更高层次的创新性研究提供有力的工具，进而提高我国知识创新的效率。实现数学的机械化是实现脑力劳动机械化的理论基础。数学机械化理论和方法的建立，是深层次的知识创新，将极大地推动相关学科的发展。数学机械化研究，不仅为数学的发展提出了一种战略构想，也将为我国信息技术的创新发挥重要作用。

数学机械化又称为机械化方法或机器证明，就是将数学的主要内容转换为计算机可接受的形式并利用计算机强大的计算功能解决数学与高新技术中的基础理论问题。主要是实现数值计算和定理证明的机械化。目前，定理机器证明是数学机械化研究的重点，同时也在人工智能的发展中有着举足轻重的影响。定理机器证明(ATP: Automated Theorem Proving)就是把一系列定理（这一系列定理可能可数，也可能不可数）当作一个整体加以考虑，建立一种统一而且确定的程序，使得此系列中的每一个定理，只要按照这个程序机械地、按部就班地一步一步进行下去，经过有限步后就可从数学命题的假设推断出数学命题的结论是否为真。这种统一的、确定的方法称为定理证明的机械化方法[3]。事实上，定理机器证明的实质就是把具有智能特点的推理演绎过程机械化。

用机械方法证明数学定理的构想是从十七世纪时期的数学家笛卡儿与莱布尼茨开始的。笛卡儿的解析几何学建立了几何的代数化，将无章可循的几何定理的证明按一定的步骤化为代数形式进行解决，为几何定理机器证明提供了简明的方法，也为几何问题的证明方法走上机械化道路奠定了基础。德国著名的数学家莱布尼兹通过对逻辑的研究，设计并制造出能做乘法的计算器，进而萌发了设计万能语言和造一台能自动检验数学命题正确性的通用机器——推理机器的构想。尽管他的这一宏伟设想，由于当时条件的限制而未能实现，但对后来兴起的布尔代数和数理逻辑却起到了促进和完善作用，正是沿着这一方向，通过人们的努力，

才形成了定理机器证明的逻辑方法。十九世纪末，希尔伯特(Hilbert)等人创立并发展了数理逻辑，又为机器证明定理提供了理论和方法。

二十世纪三十年代，法国数学家贺布兰特（Herbrand）从数理逻辑角度奠定了机器证明机械化的理论基础，为人工智能的开发提供了条件。到了四十年代，电子计算机的问世为数学定理机器证明提供了设备条件，机器证明开始崛起。波兰的著名逻辑学家、数学家塔斯基（A. Tarski）在1948年的经典著作《初等代数和几何的判定法》中，证明了计算机可以判断所有初等代数和初等几何范围内的定理证明，并且给出了机器证明的具体方法，这为人类开始机器证明方面的可行性研究树立了信心。五十年代中期，美国开始尝试利用计算机来证明数学定理。1956年可算为人类历史上计算机证明以至于人工智能研究的开端，美国卡尔基大学——兰德公司协作组的成员纽厄尔、西蒙和肖乌等人通过研究证明定理的心理过程，创建了机器证明的启发式搜索法，编制了一个“逻辑理论机”程序（简称LT），成功证明了罗素（B.A.W.Russell）和淮特海（A.N. Whitehead）所著的《数学原理》第二章52条定理中的38条。这是机器证明史上第一项奠基性的突破。

五十年代末，美籍华人、美国洛克菲勒大学的数理逻辑学家、著名教授王浩发明了王浩算法，使得机器证明开始规范化。1959年，他在IBM704计算机上仅仅用了9分钟就证明了罗素（B.A.W.Russell）和淮特海（A.N. Whitehead）所著的《数学原理》中的三百五十多个命题，并明确地提出了“走向数学的机械化”的口号，他是数学史上的第一人，当时就在数学与数理逻辑界引起轰动。1965年，美国数学家鲁宾逊（J.A. Robinson）提出了“归结原理”，这是用逻辑方法证明数学定理的一个重要进展[4]。到七十年代，机器证明获得重大的新进展。1976年美国伊利诺斯大学的哈肯（W. Haken）、科奇（J. Koch）和阿佩尔（K. Appel）在三台高速计算机上花了一千二百小时，完成了构造可约构形的不可避免集的工作，证明了一百二十四年未能解决的“四色问题”（德国数学家麦比乌斯于1840年提出了著名的四色猜想：“如果只用四种颜色能否对球面上或平面上的任何地图着色。”）。这表明利用计算机把人类思维中的逻辑推理能力推进到前所未有的高度的可能性。

七十年代，我国也参与到数学机械化这个新兴领域的研究中，并作出了巨大的贡献。1977年，我国著名数学家吴文俊院士在《中国科学》上发表了《初等几何判定问题与机械化证明》一文，文中提出了几何定理的机器证明的新方法，从而揭开了机器证明几何定理的新的一页。他首次在计算机上证明了一大批很不简单的初等几何定理，开创了从公理化到机械化的新路，在国际上被誉为“吴方法”，在吴方法的影响下，美国等十几个国家的数学家、科学家先后发表了数百篇这方面的论文[5]。1985年，吴文俊院士的《关于代数方程组的零点》的发表标志着求

解多项式方程组的吴文俊消元法(或“吴方法”)正式建立。与国际上流行的代数理想论不同，吴文俊明确提出了具有中国特色的、以多项式零点集为基本点的学术路线。“吴方法”的确立与广泛应用，复兴了国际数学机械化的研究。

我国许多的数学家在吴文俊数学机械化思想方法的启示下，在数学机械化领域也作出了很多卓有成效的工作。1999 年，我国数学家杨路创立了降维算法，并研发出实现这种算法的数学软件。他用这种软件证明了两千多个不等式，这是几何定理机械化证明领域的又一重大突破。2000 年，张景中院士研制出第三代智能数学平台软件，又称为“Z+Z”智能教育平台，开创了世界智能性数学教学软件的先河。

如今，随着数学机械化研究的深入发展，人们已经可以根据机械化方法创建各种机器语言来编制程序，并能在计算机上进行证明。短短几十年内，这一理论不仅在不等式的机器证明、几何定理的机器证明、微分几何、方程组求解、力学、理论物理等新领域得到成功的应用，而且也为信息处理、曲面造型、数控技术、机器人学、计算机图形与视觉等高科技领域提供了有力工具，并取得了一系列优秀成果。近年来上述理论又对三角函数、双曲函数等超越函数公式实现了机械化的定理证明。同时在集合论、分析拓扑学以及递归函数等方面都使机器证明获得了成功的应用。所以，随着电子计算机运行速度的不断提高，以及计算机证明理论研究的逐步深入，将会开辟机器证明的新时代。机器证明实现了用计算机来完成数学命题的证明，成为数学机械化研究的一个重要组成部分，是现代人工智能发展的一个重要方向。

迄今为止，全世界已有三百多种可以用来进行定理机器证明的语言系统[6]，其中有十五种数学定理证明系统受到大多数专家及北美数学知识管理组织（MKM）协会的认可，比如Automath系统[7]、Mizar系统[8,9]、NuPRL系统[10]和Theorema系统[11]等。这些语言系统各自的创建初衷不同，却有着一个共同的特征，就是人类使用机器语言来书写、计算、推理和证明文本性质的数学问题，并使用计算机来验证其正确性。近年来，ACL2[12]和PVS[13] 也越来越受人们的关注。荷兰著名的数学家F. Wiedijk在“Comparing mathematical provers”[14]中主要从它们知识库的大小、逻辑性表示的强弱和自动智能的水平三个方面比较Mizar、PVS、Alfa/Agda、ACL2、HOL、Coq、Otter/Ivy、Lego、Metamath、Theorema、NuPRL、Isabelle/Isar、PhoX、IMPS、Ωmega这十五个数学定理机器证明系统。此外还比较了十五个证明系统的数学化程度和自动化水平。F. Wiedijk于2006年在“The Seventeen Provers of the World” [15]中又增加了B Method系统（由Dominique Cansell创建）和Minlog系统（由Helmut Schwichtenberg创建）这两个重要的数学定理证明系统。在以上十七个系统中，HOL、Mizar、NuPRL、B Method、 PVS、Coq、Isabelle/Isar、IMPS、Minlog这9个系统均需要庞大的数学领域知识库的支持。由此可以看出，数学定理的机器证明需要有庞大的数学知识库来支持。其中，Mizar具有庞大的数学领域知识库Mizar Mathematical Library(简称MML)。

## 1.2 Mizar系统的历史和发展现状

Mizar 系统是一个利用计算机把所要处理的项目进行形式化的系统。它是由波兰华沙大学的教授 Andrzej Trybulec 组织的 Mizar 协会领导[16]的，到目前为止已经有三十多年的发展历史。最初的 Mizar 系统并不是用来书写和推证数学定理证明，而是为了利用创建的一个软件环境来帮助数学家们撰写和验证数学论文，因此发展的软件环境一般是用来书写传统的数学论文的，古典数学与集合理论也就相应的成为软件环境发展的基础[17]。

如今的Mizar系统是一个用来构建Mizar数学知识库的自动推理校验系统，拥有庞大的数学数据库(Mizar Mathematical Library，简称MML)，同时此数学数据库也是期刊Formalized Mathematics的on-line版本，在互联网上即可随时查阅。

1973 年 11 月 14 日在波兰华沙大学图书馆学和科学信息学院的一次研讨会上，Mizar 系统第一次由 Andrzej Trybulec 教授完整地提出来。他提出了设计建立一种可以用来撰写数学论文的计算机语言与它的可实现性，并提出此语言在功能上的特点[18]：

1. 能够进行存储和编译，即所写文章可以存储于计算机中，并可将其翻译成数学语言。
2. 语言简洁易懂，格式匀称整齐。
3. 具备支撑整个数学领域自动化信息系统的基本要素。
4. 可检测错误、查证参考文献、删除多余重复定理。
5. 可根据有关定理的证明进行一般的教学活动。
6. 自动排版的实现。

在 Mizar 系统发展的初期，专家们更注重系统语言的编辑功能，而并非系统的证明校验功能。1974 年秋天，由 Andrzej Trybulec、Krzysztof Lebkowski 和 Roman Matuszewski 设计的用于谓词演算验证的 ODRA-1204 系统是第一个可以运行的 Mizar 系统。尽管当时的系统十分脆弱，而且运行时间较长，但最终还是实现了完全正确的语义结构和语法分析。

1975年6月，Andrzej Trybulec教授第一次以文字的形式展示了他所创建的机器证明语言——Mizar语言，并于同年11月发表了一篇关于Mizar-PC的文章。文中介绍了可以用Jaśkowski自然演绎推理方法撰写，用逻辑谓词验算的Mizar语言的证明过程。那时的Mizar系统已经可以用（<u>begin</u> … <u>end</u>）、(<u>proof</u> … <u>end</u>)、(… <u>by</u> …)、 (let ...)、(thus ... or hence...)这样的语义词来编写下面这样的证明过程了。

```
begin
((p ⊃ q) ∧ (q ⊃ r)) ⊃ (p ⊃ r)
proof
    let   A: (p ⊃ q) ∧ (q ⊃ r) ;
    then  B: p ⊃ q ;
          C: q ⊃ r by A ;
    let      p ;
    then     q by B ;
    hence    r by C
end
end
```

之后，在华沙大学的两个学院，Mizar-PC被应用于命题逻辑课程的教学中。

1977年，Mizar系统的发展进程虽然艰难，但仍然完成了由Mizar-PC到Mizar-QC/1204（QC即Quantifier Calculus的缩写）的进化，并实现了系统发展成为Mizar-QC/6000的目标。Mizar系统已经可以完成部分简单定理的证明，如关于集合的交、并、补和格论中的偏序集的定理等，并开始研究几何学定理的证明。

1978年，Mizar系统在语义与语法上做了大量地充实、改进。在语义方面定义了谓词、结构、预声明、绝对相等的概念，系统升级为Mizar-MS（即Multi Sorted的缩写），很多命题已不再需要讨论就能完成系统的自行验证了

1978--1979年，在Stanisław Żukowski，Czesław Byliński，Andrzej Trybulec，Roman Matuszewski，Piotr Rudnicki，Elżbieta Ramm，Edmund Woronowicz七位教授的努力下，Mizar系统实现了Mizar-FC的升级。

1981年，Mizar系统升级为Mizar-2阶段。为了使文章的内容更加规范、清晰，Mizar文章中出现了环境部部分。Mizar系统在IBM、和UNIX和ODRA-1305 (ICL 1900)等多种不同的机器上试验运行，充分验证了Mizar语法的正确性及逻辑的合理性。

1982年，Mizar-MSE（Multi-Sorted predicate calculus with Equality的缩写）阶段的Mizar系统升级成功，此系统次年8月的华沙国际数学家会议上深受瞩目。同年，系统接着完成了到Mizar-3（Mizar-2的扩充版本）的进一步完善。Mizar系统在1983--1984年和1986--1988年分别实现了Mizar-HPF和Mizar-4两次重大的突破。现今所用版本就是由Mizar-4（Mizar-3的再次升级）演化而来的[19]。

1988--1989年是Mizar系统蜕变的重要时期。1988年，Mizar-4被称之为PC-Mizar，并在PC机上成功运行。1989年初，Mizar数学数据库Mizar Mathematical Library（MML）开始正式使用。其中，第一篇标准的Mizar文章于1月6日被收录于MML中。同年4--5月，Trybulec教授在加拿大的艾特蒙顿开始了新一篇Mizar文章的撰写，并着手将Mizar系统中加入$\TeX$，加强了Mizar文章的可读性。随着Mizar数据库的不断扩大，Mizar系统也在进行更新升级。1990年，Mizar系统的文本期

刊《Formalized Mathematics》开始正式出版发行。1995--1997年，《Formalized Mathematics》得到美国Office of Naval Research的资金支持[19]。到2000年，Mizar数学百科全书问世。第二年，Mizar协会的教授Grzegorz Bancerek将MML Query增加到Mizar系统中，作为在语义方面进行问题查询的浏览工具。2002年，Mizar数学百科全书EMM（Encyclopedia of Mathematics in Mizar）完成[19]。如今，Mizar系统已经升级为7.11.01版本，MML中也已收录了一千余篇文章，二万多个数学定义，四十多万条数学定理[20]。

## 1.3 课题研究的目的和意义

本课题的研究是建立在对数学定理的机器证明和数学定理自动推理系统Mizar及其语义、语法的深刻理解之上的，在Mizar系统中利用其语言及庞大的数学数据库（MML），对特殊函数差分差商、特殊函数积分、函数正交性与函数范数做进一步研究和应用。通过对数学机械化与Mizar系统的探索研究，为分析学、规划论、数值逼近等方面问题在Mizar系统中的解决实现奠定理论与实践基础。

从定理自动证明的发展过程中产生和发展起来的Mizar系统，是一个既可撰写数学文章又可对其进行检验的软件，绝大多数数学学科中的问题都能用Mizar语言来证明。除了本身重要之外，Mizar系统还为研究广泛领域的机械化数学和数学机械化研究提供了新方法、新工具、新思路。日本数学家和计算机科学专家Yatsuka Nakamura教授利用Mizar系统的自动推理完成了Jordan曲线定理的证明，并应用在人工智能方面，使得Mizar参与实现了微型机器人、音像识别系统、自动化控制、数字信号传输等多方面的成果研究。

现在的Mizar系统已经从单独定理证明发展到多学科多方面的交叉应用，其发展主要包括以下内容：

1．Mizar语言的开发；

2．Mizar数据库的扩充；

3．Mizar自动推理能力的提高；

4．Mizar翻译形式的多样化处理。

通过对这些方面的研究与完善，Mizar语言的理解能力逐步提升，MML的知识储备得到扩充，Mizar系统的逻辑推理和校验纠错的能力得到增强，翻译机制更加完善，与其他机器证明工具的交叉研究和相互转化不断加强。Mizar的发展需要庞大的数学领域知识库（MML）支持，数学学科的发展也将推动数学定理机器证明系统——Mizar的发展。随着社会进步与计算机信息产业的发展，Mizar系统将会在更多领域得到应用。

## 1.4 课题研究的主要内容

Mizar系统从1989年建立以来，成功的完成了多次升级，同时数据库Mizar Mathematical Library（MML）也不断进行着充实和改进。Mizar数据库MML涉及到数学领域中的多个学科，如数学分析、代数学、几何学、拓扑学、数论、图论、集合论等。但关于特殊函数差分差商、特殊函数积分及函数正交性与函数范数方面的定理鲜有涉及。

本文的主要内容如下：

1．基于Mizar中差分差商的定义与性质，完成一次函数、二次函数、反比例函数、三角函数等特殊函数的差分差商在Mizar系统中的实现。

2．根据指数函数、三角函数、幂函数、双曲函数、无理函数、反三角函数等特殊函数的微分公式及可积性，运用牛顿莱布尼茨公式，在Mizar系统中实现多种特殊函数的积分。

3．为了实现以$\rho(x)=1$为权函数时函数正交性的Mizar推理，首先在Mizar系统中定义了内积，并论证了其性质。再使用Mizar语言给出函数正交性与函数范数的定义，并描述了其性质，同时完成以上内容的Mizar系统证明。

本文主要通过研究特殊函数差分差商、特殊函数积分及函数正交性与函数范数，完成相关定理的Mizar论证，并实现有关数学理论的Mizar转化。

# 2 Mizar语言

## 2.1 Mizar语言

Mizar语言是一种数学逻辑推理演绎的计算机语言，它包括英文字母、数字和各类标号在内的ASCII码，这些基本符号是依据常用的数学符号与一般的ASCII码相结合的原则来定义的。Mizar语言是计算机可编译、人类可识别的一种程序化语言。其主要按如下学科分类[21]：数的集合、集合论、一般函数(或关系)、算术运算、特殊常量、复数及基本函数、极限和序列(级数)、逻辑、微积分、几何学、概率、向量和矩阵。

常用的Mizar语言词汇和基本的语法句型主要有：reserve（数据属性预声明）、let（“任意取定”词)、consider（“存在取定”词)、reconsider…as…（数据属性转换词)、for…holds…（全称命题)、implies（直接推导词)、suppose（分类讨论引导词)、assume（假设词)、given（存在条件引导词）、now...thus...（扩散陈述命题）、such that与st（条件引导词)、then（因果连接词)、by（简单依据引导词)、from（结构依据引导词)、thus与hence（结论引导词）等等。

任何一个命题大多是由这些词语连接命题的表述和证明的。语法词汇也基本蕴含在各个具体的证明方法中，如分类讨论法、数学归纳法、反证法、同一性证明法等等。

## 2.2 Mizar文章结构

Mizar文章是使用Mizar语言进行定义、计算、推理演绎以及数学定理证明而形成的文章，利用Mizar系统提供的软件可以在计算机上对其进行检验和优化。通过系统验证准确无误的Mizar论文将被发表在期刊杂志《*Formalized Mathematics*》上，并收录到Mizar数学数据库(MML)中。每一篇Mizar文章都是文本文件，可以在任何一个文本编辑软件中进行编写编辑。Mizar文章包括两部分——环境部(the environment)与正文部分(main section)。环境部以“environ”开始，以分号结束。正文部分以“begin”开始，以分号结束。若正文部分的内容较多，还可根据需要用“begin”将正文部分分成几个小专题，每一个小专题都同样以begin开始，以分号结束(如图2-1所示)。

```
environ
    The environ section ;
begin
    The main section 1;
begin
    The main section 2;
        ..................
begin
    The main section n ;
```

图2-1 Mizar文章的一般结构

Fig.2-1 the frame of Mizar articles

环境部（“environ”部分）主要对论文中所引用的词汇、符号、定理、初定义及结构的出处进行声明(如图 2-2 所示)。“vocabularies” 部分是文章引用的词汇列表，同时也表明了算符词汇的运算优先级，如数学运算符的优先级就是通过文章名排列的先后顺序体现出来的。这就意味着作者可以使用数据库里已定义好的数百个词汇，也可以根据需要自己定义新的词汇。“signature” 部分(包括 notations，constructors，registrations，requirements 四部分)通知 Mizar 处理器这篇文章可以使用此处列出的文章里的符号。任一篇文章都可以根据需要，使用词汇列表中的词汇来生成各种形式的新的 Mizar 表达式。“definitions”、“theorems”和“schemes”就是告诉 Mizar 处理器可以使用这些列表文章中的定义、定理和模式等。

在Mizar文章中若要使用系统中已存在的定义或定理，必须将定义或定理的出处分别添加到环境部的相应分类中。如果在文章中定义了新概念、新符号或新结构等，同样需要将本篇文章的缩写名写入环境部中相应的位置。缩写名可由自己命名，便于在Mizar系统中的调用与检验。

| | | |
|---|---|---|
| 词汇 | vocabularies | $a_1, a_2, \cdots, a_{n_1}$; |
| 符号 signature | notations | $b_1, b_2, \cdots, b_{n_2}$; |
| | constructors | $c_1, c_2, \cdots, c_{n_3}$; |
| | registrations | $d_1, d_2, \cdots, d_{n_4}$; |
| | requirements | $e_1, e_2, \cdots, e_{n_5}$; |
| 初定义 | definitions | $f_1, f_2, \cdots, f_{n_6}$; |
| 定理 | theorems | $g_1, g_2, \cdots, g_{n_7}$; |
| 结构 | schemes | $h_1, h_2, \cdots, h_{n_8}$; |

图 2-2 环境部引用分类词汇

Fig.2-2 the vocabulary of environment

其中$a_1, a_2, \cdots, a_{n_1}, \cdots, h_1, h_2, \cdots, h_{n8}$,表示被引用的Mizar文章名。

Mizar 文章的正文部分一般包含以下几个模块：

1．预声明模块。用来声明变量（标识符）的类型。

2．定义模块。用来定义或者重新定义 Mizar 的语法构造，包括 functions（数学算子的构造）、predicates（公式构造）和 modes（类型构造）等。

3．结构定义模块。用来定义新结构。一个由各种域构成的实体就是一个结构。

4．定理的证明。用来对数学命题进行证明。借助 Mizar 语言编写的数学命题通过验证无误后即可被其它定理或文章引用。

5．辅助信息。包含了一些本篇文章可以使用的引论或引理，但这些内容不会被收录到数据库中，其它文章也不能引用。

Mizar文章的正文部分主要由定义definitions和定理theorems构成。定理的证明以“proof”开头，以“end;”结束，二者成对出现。如果证明过程中需要证明一些小命题，那么这些小命题的证明也采取同样的格式，如图2-3：

```
theorem   A formula that is needs to be demonstrated
          (需要证明的定理或公式)
  proof  ...
    A little proposition that is needs to be demonstrated
          (需要嵌入的小命题)
    proof  ...
    end;
    ...
    thus(hence) ...;
  end;
```

图2-3 定理证明的一般结构

Fig.2-3 the frame of the demonstration

特别说明：Mizar文章中，“::”（双冒号）表示对问题的评论，检验系统不对其后内容的正确性与合理性进行验证。“@”是免检符号，放在“proof”前，表示校验系统对该“proof”和与它成对出现的“end”之间的证明过程不予检验，默认是正确的。

## 2.3 Mizar语言的基本表达符号与基本语法词汇

### 2.3.1 Mizar语言的基本表达符号

Mizar语言中的符号基于纯粹的数学符号，这些符号的英文表述和一般的ASCII字符相结合来进行定义与编写。举例如下表：

| 普通数学命题中的逻辑符号 | ∀ | ∃ | ≥ | ≤ | ≠ | ∩ | ∪ | ∈ | ∅ | ⊆ | ⇔ |
|---|---|---|---|---|---|---|---|---|---|---|---|
| Mizar 语言系统中的逻辑符号 | for | ex | >= | <= | <> | /\ | \/ | in | {} | C= | iff |

表2-1 普通数学命题和Mizar语言系统中的逻辑符号

### 2.3.2 Mizar语言的基本语法词汇

Mizar语言中的语法包括定义、证明和撰写一篇Mizar文章的必要组成。下面按Mizar文章的正文部分撰写的一般顺序做一下介绍。

1、数据属性预声明reserve

在标志文章正文部分开始的“begin”之后，变量首先要用reserve声明数据类型后才能使用。例如：

begin

reserve Z for set;

reserve x, y for Real;

reserve f for PartFunc of REAL,REAL;

用 reserve 做了如上的声明，Mizar 系统在处理论文时候，就会默认 Z 为集合，x, y 为实数，f 为实数域到实数域上的函数。而对于正文部分没有预先声明的变量，每次引用或提及时都要格外指明，否则 Mizar 系统不承认其存在的合理性。例如：

ex m being Element of NAT st m>n by XREAL_1:3;

因为 m 没有在 reserve 中预先声明数据类型，所以每次提及时需要说明 being Element of NAT（自然数集的元素）。

2、推导词implies

implies 用于定理的表述，是条件与结论之间的连接词。如：A$\Rightarrow$B可表示为

A implies B

这里不要求A、B间有直接的因果关系。

3、全称命题for … holds …

（1）一般的定理表述

for x being set holds A(x) (任给x，能推出A(x))

（2）含条件的定理表述

for x being set st B(x) holds A(x) (任给x，满足B(x)，能推出A(x))

（3）含存在性的定理表述

for x being set st B(x) holds ex y st A(x,y) (任给x，满足B(x)，可以推出存在y，使得A(x,y)成立)。

4、假设词assume

assume用于引入命题条件，后面紧接假设条件的表述。例如：

assume x in dom f（假设x在f的定义域中）。

5、简单依据引导词by和结构依据引导词from

在Mizar系统中，任何命题的推导都要有严格的结论依据，一般情况下直接结论依据由by引导。这些依据可以是已收录于Mizar数据库的文章中的定义或定理（收录在MML中的文章，系统会自动生成后缀为.abs的论文摘要，摘要中包含着论文中出现的定义和定理，并自动给每个定义和定理自动排序，生成便于系统校验和读者查阅的ASCII码简写名），也可以是正在书写论文中的已证明成立的命

题。例如：

assume A1: x in X;

A2: X c= Y;

x in Y by A1, A2, TARSKI:def 3;

by 引导此命题推导的依据，其中 A1, A2 为 Mizar 论文中的标号，指代着文章中成立或证明过的命题，为本地引用；还有一个依据为 TARSKI:def 3，为数据库引用，即引用了 TARSKI 这篇文章中的第三个定义。

from 同 by 的功能一致，区别在于 from 引导的结论依据必须是结构模式 scheme。例如：

consider s1 such that for n holds s1.n = F(n) from SEQ_1:sch 1

此论述依据的是SEQ_1文章中的结构模式sch 1，故用from引导。

6、因果连接词then、结论引导词thus和含因果关系的结论引导词hence

then相当于连接词“所以”，用来连接两个有直接因果关系的命题。thus用来引导命题结论，thus以上是对命题的论证部分，其后紧跟着成立的结论。而hence是then与thus的合成词，兼具有then和thus的功能（数学运算模拟为：hence = then + thus），用于推出完整性结论。例如：

A1: A = B;

A2: B = C;

thus A = C by A1,A2;

如果用hence，上述证明语句也可以如下表示：

A1: A = B;

B = C;

hence A = C by A1;

7、条件引导词such that和st

条件引导词用来引导假设或命题中提及的变量所满足的条件，在定义定理的表述中用st形式，在命题的证明过程中用such that形式。

8、“任意取定”词let和“存在取定”词consider

在Mizar语言中表示取定意思的词汇有两个：let和consider。let一般用在全称命题“for ... holds ...”的证明过程中，与for引导的条件命题相对应，表示在满足假设条件的变量中，任意取定一个证明论证，侧重于由任意性推导命题成立。consider与“ex”引导的条件命题相对应，是针对存在性的条件命题进行的假设，侧重于取定事实存在的某个变量，借助其满足的条件推导命题成立。

9、数据属性转换reconsider … as …

在论文的书写中，改变变量的数据类型是不可避免的。reconsider...as...用来改变数据类型。例如：

reconsider y1 = y as Real by ...

此句中用新的变量y1代替y，实现了变量y的属性到Real的转变。

10、分类讨论引导词case

Mizar中针对需要分类讨论的问题，先给出分类的合理说明，然后对每一种具体情况，使用case来分别引导进行推理证明，再综合所有分情况讨论得到最终结论的成立。

11、存在条件引用词given

given 引导存在性假设条件，是 now 和 consider 的合成词，兼有假设(now)和取定(consider)两个词的意思，数学运算模拟可表示为：given = now + consider。例如：

```
theorem   (ex x st A[x]) implies for y st B[y] holds C[y]
proof
  given x such that A[x];       :: given引导存在性假设条件——ex x st A[x]
  ……
  let y;                        :: let与命题表述中for引导的条件相对应
  assume B[y];
  ……
  thus C[y];
end;
```

12、扩散陈述命题now...thus...

以now开始扩散陈述命题的证明方法，一般要求now之前没有then、thus或hence等词语，另外在命题证明结束时thus后跟的结论必须明确写出，不能用thesis替代。例如：

```
now
  assumeA;
  ……
  take x;
  ……                  ⇒ 即证明了命题：A implies es x st B
  thus B;             ::B指代的结论需完整表述，而不能用thesis代替
 end;
```

## 2.4 Mizar系统的证明方法

Mizar是基于Jaśkowski自然演绎推理的古典逻辑构建而成的，拥有符合古典逻辑推理的基本证明方法。

### 2.4.1 一般表述与证明格式

Mizar语言中对一般命题的表述与传统数学语言的命题陈述类似，主要分为七种命题：

1、简单命题

```
theorem   P[X]                    :: P[X]是关于 X 的一个命题
proof
   ......
  thus P[X];
end;
```

2、与命题（&）

```
theorem   P[X] & Q[Y]             :: P[X]是关于 X 的一个命题,
proof                             :: Q[Y]是关于 Y 的一个命题
   ......
  thus P[X];
  thus Q[Y];
end;
```

3、推导命题（implies）

```
theorem   P[X] implies Q[X]       :: P[X]⇒Q[X]
proof
  assume P[X];
   ......
  thus Q[X];
end;
```

4、等价命题（iff）

```
theorem   P[X] iff Q[X]           :: P[X]⇔Q[X]
proof
   ......
  thus P[X] implies Q[X];
   ......
  thus Q[X] implies P[X];
end;
```

或者

```
theorem   P[X] iff Q[X]
proof
  hereby
  assume P[X];
   ......
  thus Q[X];
```

```
  end;                              ::这个 end 与 hereby 相对应，成对出现
 assume Q[X];
   ......
  thus P[X];
 end;
```

5、或命题（**or**）

```
 theorem   P[X] or Q[X]
 proof
   assume not P[X];
    ......
   thus Q[X];
 end;
```

6、全称命题（**for** … **holds** … ）

```
 theorem   for X being set holds P[X]
 proof
   let X be set;
    ......
   thus P[X];
 end;
```

7、存在命题（**ex** … **st** … ）

```
 theorem   ex X being set st P[X]
 proof
   take X;
    ......
   thus P[X];
 end;
```

### 2.4.2 基本证明方法

1、直接法

直接法即从命题的条件出发，通过正确合理的推导过程得出命题的结论。如命题“若a，则b”：

```
 a implies b
 proof
   assume a;              :: a是条件
     ......
   thus b;                :: 证明b成立
  hence thesis;
```

```
end;
```

2、反证法

反证法，又称矛盾法、无理法，即假设命题结论的否命题成立，通过正确合理的推导过程，得出矛盾。“contradiction”的使用是反证法的标志。如命题“若a，则b”：

```
a implies b
proof
  assume a;
  assume not b;              :: 假设非b成立
    ......
  thus contradiction;        :: 证明矛盾，说明a与非b不能同时成立
end;
```

### 2.4.3 特殊命题的证明方法

1、证明与命题

拆分法，就是将命题拆成多个小命题，通过对小命题逐个证明，以完成整个命题的正确性证明。如命题“a & b & c”：

```
a & b & c
proof
    ......
  thus a;
    ......
  thus b;
    ......
  thus c;
    ......
end;
```

2、证明等价命题

双向法，就是将等价命题看作“左推右”和“右推左”两个方向的单命题，再顺次（先左到右，后右到左）采用基本方法中的直接法或反证法进行推理证明。如命题“a iff b”，其中蕴含两个命题“a implies b”和“b implies a”：

```
a iff b
proof
  assume a;
    ......
  thus b;                    :: 直接法证明 a => b
```

```
    assume b;
      ......
    thus a;                    :: 证明 b => a，因此 a <=> b
  end;
```

3、证明或命题

单一否定法，就是否定其一的结论，通过合理推理，完成肯定另一个命题的证明。

如命题“a or b”：

```
  a or b
  proof
    assume not a;
      ......
    thus b;                    :: 直接法证明a或b
  end;
```

或者a or b

```
  proof
    assume that A:not a and B:not b;
      ......
    thus contradiction;        :: 无理法证明a或b
  end;
```

4、证明存在性命题

对于命题的存在性证明，一般用“take”作总结证明。如命题“ex x st P[x]”：

```
  ex x st P[x]
  proof
      ......
    L:P[a];                    :: 至此证明∃a,满足P(a)
    take a;                    :: 取a，即可thus thesis by L;
  end;
```

实际上，在Mizar中所遇到的定理的表述往往是以上几种的综合，所以定理的证明也是多种证明方法的综合。

## 2.5 Mizar语言中的定义（Definition）

Mizar语言系统中，对定义的要求十分严格，定义不但要严密合理，而且要分类明确。Mizar语言中的定义主要包括模式定义(modes)、谓词定义(predicates)、算子定义(fuctors)、属性定义(attributes)、cluster定义、结构定义(structures)。对于不

同分类的定义，系统对是否需要从存在性和一致性两方面给出逻辑证明做出了确切规定，如表2-2所示[22]，“ο”表示需要证明，无“ο”表示无须证明。

| | Existence | Uniqueness |
|---|---|---|
| Mode（模式） | ο | |
| Pred（谓词） | | |
| Functor（算子） | ο | ο |
| Attr（属性） | | |
| Cluster | ο | ο |

表2-2 定义证明分类

有新定义的Mizar文章都要建立一个从属文章的词汇文档，该文档与文章同名，且以.voc为后缀。收录在MML的文章中的所有新定义都会被收录到词汇数据库中（.voc文件），因此新定义的术语不能与Mizar数据库中已存在的数学术语重复。对于不同类型的定义，词汇文档的书写形式也是不同的。在新定义的符号或术语前面，需用一个大写字母表明其类型，具体表示见表2-3[22]。

| Symbols（代表字母） | Things that are defined（定义类型） |
|---|---|
| M | Mode（模式） |
| O | Functor（算子） |
| R | Predicate（谓词） |
| K | Left functor bracket（左括号） |
| L | Right functor bracket（右括号） |
| G | Structure（结构） |
| U | Selector（函子） |
| V | Attribute（属性） |

表2-3 定义类型的代表字母表

voc文件的创建规则及顺序是：首先填写表示类型的一个大写字母，其后紧接新定义的词汇名，它们之间不可添加空格或其他字符。特别地，对于新定义的算子，在类型字母O加新定义词汇名后，还可添加空格和1到255之间的一个数字，这样添加后的数字表示运算优先级，若缺省则表示优先级为64。下表2-4是Mizar文章FUNCT_1的词汇表——FUNCT_1.voc，在函数正交性与函数范数的Mizar实现中将举例说明其他类型的定义方式。

| FUNCT_1.VOC |
|---|
| VFunction-like |
| MFunction |
| O. 100 |
| Oid 128 |
| Vone-to-one |

表2-4 FUNCT_1.VOC的创建形式

## 2.6 Mizar系统的逻辑推理检验及优化命令

对于Mizar论文书写过程中出现的错误，Mizar系统不但用不同的错误标号将其标识，而且在论文的最后对错误出现的原因给出解释，从而便于作者及时发现与修正。下面介绍一下Mizar系统检验论文的常用命令。

1. mizf　　检查推理过程正确与否。（在文章撰写过程中随时都可运行mizf命令，以校验Mizar论文表达或逻辑推理的正确性，是论文书写过程中使用频率最高的命令。）

2. errflag　　错误显示命令。通知Mizar系统对论文出现的错误添加标记。

3. notepad　　用记事本方式将论文打开。

4. listvoc　　显示某篇文章的词汇文件。

5. checkovc　　查找数学术语或符号的出处。

此外在论文完成之后，可运行Mizar系统自带的其他命令程序对文章进行完善和优化。主要有以下八个命令：

1. relprem　　检验文章中多余的引用
2. chklab　　检验文章中多余的标号
3. inacc　　检验文章中多余的论证步骤
4. trivdemo　　检验文章中多余的证明过程
5.relinfer　　检验文章证明过程中可简化的步骤，合并繁琐的表达
6. reliters　　检验文章中多余的运算步骤
7. irrvoc　　检验环境部中多余的词汇
8. irrths　　检验环境部中多余的定理

其中，irrvoc命令与mizf命令交替运行，可以完成对环境部中notations和constructors两部分的检验。另外，irrths命令也可优化环境部中theorems和schemes部分的内容。以上8个优化命令必须与错误显示命令errflag交替运行，才能完成检验过程。

## 2.7 Mizar语言系统的更新升级

Mizar语言系统和数学数据库的版本在不断的更新升级，最新的版本可以在官方网站：http://www.mizar.org/和http://markun.cs.shinshu-u.ac.jp//pub/mizar/上查找得到。

# 3 特殊函数差分差商的Mizar实现

Mizar 系统既能完成推理研究与证明，也具有一般的计算功能，特殊函数差分差商的 Mizar 实现就很好的体现了这一点。

在 Mizar 数据库（MML）中，差分、差商的定义与相关性质已经基本完备，但没有研究分析特殊函数的差分差商，也没有给出其公式。本章主要研究如何将以上内容以 Mizar 系统表述和推证。

## 3.1 特殊函数的差分差商

函数的有限差分差商在运筹学、数值分析、概率统计和计算机软件等科学中被广泛地应用，例如：微分方程的数值解法、曲线的数值拟合、动力控制、数字图像处理等等[23]。本章只选取部分定理给出Mizar表述和证明方法。所有的证明收录于MML中的DIFF_2.miz[24]中，发表于2008年的Mizar期刊 ***Formalized Mathematics***。

## 3.2 特殊函数差分差商的Mizar实现

Mizar 系统中，已经包含差分、低阶差商的概念与相关性质。在此基础上，通过近一步分析研究，可以给出特殊函数的差分与差商公式。

### 3.2.1 特殊函数差分差商的环境部设置与变量声明

Mizar论文的开始是环境部，在论文撰写初期，不能确定环境部出现的文章是否需要引用，所以刚开始建立的环境部并不是完全固定的，要根据情况对其进行添加或删减。在此给出论文撰写完毕后环境部的最终形式，如下：

**environ**

**vocabularies** INCPROJ, ARYTM_1, SEQM_3, SEQ_1, FUNCT_1, ARYTM_3, SQUARE_1, SIN_COS, DIFF_1, ARYTM, RELAT_1, JGRAPH_2, SIN_COS4, VALUED_0;

**notations** SQUARE_1, ORDINAL1, XCMPLX_0, ZFMISC_1, REAL_1, NAT_1, NUMBERS, FUNCT_1, FUNCT_2, VALUED_1, SEQ_1, SEQFUNC, RELAT_1, SIN_COS, DIFF_1, XREAL_0, VALUED_0, JORDAN16, SIN_COS4, FDIFF_9;

**constructors** SQUARE_1, REAL_1, NAT_1, DIFF_1, PARTFUN1, PARTFUN3, SIN_COS, SIN_COS4, JORDAN16, SEQ_1, FDIFF_9, SEQFUNC;

**registrations** XREAL_0, RELSET_1, MEMBERED, FUNCT_2, NUMBERS,

```
VALUED_0, VALUED_1;
    requirements NUMERALS, SUBSET, BOOLE, ARITHM;
```

正文部分以“begin”开始，接着声明文章中需要使用变量的数据类型，如下：

```
begin
reserve h,x0,x1,x2,x3,x,a,b,c,k,y,z for Real,
      f for Function of REAL,REAL;
```

3.2.2 特殊函数差分差商的Mizar表述与实现

我们将讨论的特殊函数有：一次函数、二次函数、反比例函数、三角函数类。下面举例说明。

一次函数形式为

$$f(x) = ax + b$$

在 Mizar 系统中此函数用 AffineMap(a,b)表示，其向前差分公式表述为：

```
theorem :: DIFF_2:24
  for x holds fD(AffineMap(a,b),h).x = a*h;
```

“theorem”后的“::DIFF_2:24”表示此定理被收录于文章 DIFF_2.miz[24]，是文章中的第 24 个定理。定理中的 x 为任意实数，fD 表示向前差分，h 表示步长，*表示乘号。以下 bD 表示向后差分，cD 表示中心差分。

向后差分公式表述为：

```
theorem :: DIFF_2:25
  for x holds bD(AffineMap(a,b),h).x = a*h;
```

中心差分公式表述为：

```
theorem :: DIFF_2:26
  for x holds cD(AffineMap(a,b),h).x = a*h;
```

一阶差商公式表述为

```
theorem :: DIFF_2:21
  x0<>x1 implies [!AffineMap(a,b),x0,x1!] = a;
```

二阶差商公式表述为：

```
theorem :: DIFF_2:22
  x0,x1,x2 are_mutually_different implies [!AffineMap(a,b),x0,x1,x2!]=0;
```

这里变量 x0,x1,x2 互不相等用 x0,x1,x2 are_mutually_different 表示。

三阶差商公式表述为：

```
theorem :: DIFF_2:23
  x0,x1,x2,x3 are_mutually_different
  implies [!AffineMap(a,b),x0,x1,x2,x3!]=0;
```

一次函数差分差商在 Mizar 论文中的表述为[24]：

(24) For every $x$ holds $(\mathrm{fD}(\mathrm{AffineMap}(a,b),h))(x) = a \cdot h$.
(25) For every $x$ holds $(\mathrm{bD}(\mathrm{AffineMap}(a,b),h))(x) = a \cdot h$.
(26) For every $x$ holds $(\mathrm{cD}(\mathrm{AffineMap}(a,b),h))(x) = a \cdot h$.
(21) If $x_0 \neq x_1$, then $\Delta(\mathrm{AffineMap}(a,b), x_0, x_1) = a$.
(22) If $x_0, x_1, x_2$ are mutually different, then $[!\,\mathrm{AffineMap}(a,b), x_0, x_1, x_2\,!] = 0$.
(23) If $x_0, x_1, x_2, x_3$ are mutually different, then $[!\,\mathrm{AffineMap}(a,b), x_0, x_1, x_2, x_3\,!] = 0$.

二次函数形式为

$$f(x) = ax^2 + bx + c$$

向前差分公式表述为：

```
theorem :: DIFF_2:31
  (for x holds f.x = a*x^2+b*x+c)
  implies for x holds fD(f,h).x = 2*a*h*x+a*h^2+b*h;
```

x^2 表示 x 的平方。

向后差分公式表述为：

```
theorem :: DIFF_2:32
  (for x holds f.x = a*x^2+b*x+c)
  implies for x holds bD(f,h).x = 2*a*h*x-a*h^2+b*h;
```

中心差分公式表述为：

```
theorem :: DIFF_2:33
  (for x holds f.x = a*x^2+b*x+c) implies
    for x holds cD(f,h).x = 2*a*h*x+b*h;
```

一阶差商公式表述为：

```
theorem :: DIFF_2:27
  (for x holds f.x = a*x^2+b*x+c)
 & x0<>x1 implies [!f,x0,x1!]=a*(x0+x1)+b;
```

二阶差商公式表述为：

```
theorem :: DIFF_2:28
  (for x holds f.x = a*x^2+b*x+c)
 & x0,x1,x2 are_mutually_different implies [!f,x0,x1,x2!]=a;
```

三阶差商公式表述为：

```
theorem :: DIFF_2:29
  (for x holds f.x = a*x^2+b*x+c)
& x0,x1,x2,x3 are_mutually_different implies [!f,x0,x1,x2,x3!]=0;
```

二次函数差分差商在 Mizar 论文中的表述为[24]：

(31) If for every $x$ holds $f(x) = a \cdot x^2 + b \cdot x + c$, then for every $x$ holds $(\mathrm{fD}(f,h))(x) = 2 \cdot a \cdot h \cdot x + a \cdot h^2 + b \cdot h$.

(32) If for every $x$ holds $f(x) = a \cdot x^2 + b \cdot x + c$, then for every $x$ holds $(\mathrm{bD}(f,h))(x) = (2 \cdot a \cdot h \cdot x - a \cdot h^2) + b \cdot h$.

(33) If for every $x$ holds $f(x) = a \cdot x^2 + b \cdot x + c$, then for every $x$ holds $(\mathrm{cD}(f,h))(x) = 2 \cdot a \cdot h \cdot x + b \cdot h$.

(27) If for every $x$ holds $f(x) = a \cdot x^2 + b \cdot x + c$ and $x_0 \neq x_1$, then $\Delta(f, x_0, x_1) = a \cdot (x_0 + x_1) + b$.

(28) If for every $x$ holds $f(x) = a \cdot x^2 + b \cdot x + c$ and $x_0$, $x_1$, $x_2$ are mutually different, then $[!f, x_0, x_1, x_2!] = a$.

(29) If for every $x$ holds $f(x) = a \cdot x^2 + b \cdot x + c$ and $x_0$, $x_1$, $x_2$, $x_3$ are mutually different, then $[!f, x_0, x_1, x_2, x_3!] = 0$.

反比例函数形式为

$$y = k/x$$

向前差分公式表述为：

```
theorem :: DIFF_2:38
  (for x holds f.x = k/x & x<>0 & x+h<>0)
  implies for x holds fD(f,h).x = (-k*h)/((x+h)*x);
```

向后差分公式表述为：

```
theorem :: DIFF_2:39
  (for x holds f.x = k/x & x<>0 & x-h<>0)
  implies for x holds bD(f,h).x = (-k*h)/((x-h)*x);
```

中心差分公式表述为：

theorem :: DIFF_2:40
  (for x holds f.x = k/x & x+h/2<>0 & x-h/2<>0)
  implies for x holds cD(f,h).x = (-k*h)/((x-h/2)*(x+h/2));

一阶差商公式表述为:
theorem :: DIFF_2:34
  (for x holds f.x = k/x) & x0<>x1 & x0<>0 & x1<>0
  implies [!f,x0,x1!] = - k/(x0*x1);

二阶差商公式表述为:
theorem :: DIFF_2:35
  (for x holds f.x = k/x) & x0<>0 & x1<>0 & x2<>0
  & x0,x1,x2 are_mutually_different implies [!f,x0,x1,x2!] = k/(x0*x1*x2);

三阶差商公式表述为:
theorem :: DIFF_2:36
  (for x holds f.x = k/x) & x0<>0 & x1<>0 & x2<>0 & x3<>0
  & x0,x1,x2,x3 are_mutually_different
  implies [!f,x0,x1,x2,x3!]=-k/(x0*x1*x2*x3);

反比例函数差分差商在 Mizar 论文中的表述为[24]:

(38) If for every $x$ holds $f(x) = \frac{k}{x}$ and $x \neq 0$ and $x + h \neq 0$, then for every $x$ holds $(\mathrm{fD}(f, h))(x) = \frac{-k \cdot h}{(x+h) \cdot x}$.

(39) If for every $x$ holds $f(x) = \frac{k}{x}$ and $x \neq 0$ and $x - h \neq 0$, then for every $x$ holds $(\mathrm{bD}(f, h))(x) = \frac{-k \cdot h}{(x-h) \cdot x}$.

(40) If for every $x$ holds $f(x) = \frac{k}{x}$ and $x + \frac{h}{2} \neq 0$ and $x - \frac{h}{2} \neq 0$, then for every $x$ holds $(\mathrm{cD}(f, h))(x) = \frac{-k \cdot h}{(x-\frac{h}{2}) \cdot (x+\frac{h}{2})}$.

(34) If for every $x$ holds $f(x) = \frac{k}{x}$ and $x_0 \neq x_1$ and $x_0 \neq 0$ and $x_1 \neq 0$, then $\Delta(f, x_0, x_1) = -\frac{k}{x_0 \cdot x_1}$.

(35) If for every $x$ holds $f(x) = \frac{k}{x}$ and $x_0 \neq 0$ and $x_1 \neq 0$ and $x_2 \neq 0$ and $x_0$, $x_1$, $x_2$ are mutually different, then $[!f, x_0, x_1, x_2!] = \frac{k}{x_0 \cdot x_1 \cdot x_2}$.

(36) Suppose for every $x$ holds $f(x) = \frac{k}{x}$ and $x_0 \neq 0$ and $x_1 \neq 0$ and $x_2 \neq 0$ and $x_3 \neq 0$ and $x_0$, $x_1$, $x_2$, $x_3$ are mutually different. Then $[!f, x_0, x_1, x_2, x_3!] = -\frac{k}{x_0 \cdot x_1 \cdot x_2 \cdot x_3}$.

三角函数类，以函数

$$y = \sin x$$

为例。其向前差分公式表述为:

```
theorem :: DIFF_2:42
  for x holds fD(sin,h).x = 2*(cos((2*x+h)/2)*sin(h/2));
```

向后差分公式表述为：

```
theorem :: DIFF_2:43
  for x holds bD(sin,h).x = 2*(cos((2*x-h)/2)*sin(h/2));
```

中心差分公式表述为：

```
theorem :: DIFF_2:44
  for x holds cD(sin,h).x = 2*(cos(x)*sin(h/2));
```

一阶差商公式表述为：

```
theorem :: DIFF_2:41
  [!sin,x0,x1!] = 2*cos((x0+x1)/2)*sin((x0-x1)/2)/(x0-x1);
```

函数 $y = \sin x$ 差分差商在 Mizar 论文中的表述为[24]：

(42) For every $x$ holds (fD(the function sin, $h$))($x$) $= 2 \cdot (\cos(\frac{2 \cdot x+h}{2}) \cdot \sin(\frac{h}{2}))$.

(43) For every $x$ holds (bD(the function sin, $h$))($x$) $= 2 \cdot (\cos(\frac{2 \cdot x-h}{2}) \cdot \sin(\frac{h}{2}))$.

(44) For every $x$ holds (cD(the function sin, $h$))($x$) $= 2 \cdot (\cos x \cdot \sin(\frac{h}{2}))$.

(41) $\Delta$(the function sin, $x_0$, $x_1$) $= \frac{2 \cdot \cos(\frac{x_0+x_1}{2}) \cdot \sin(\frac{x_0-x_1}{2})}{x_0-x_1}$.

以上特殊函数的差分差商讨论及逻辑证明过程参见文章DIFF_2[24]。

### 3.2.3 特殊函数差分差商的Mizar证明

以特殊函数

$$f(x) = \sin^2 x \cos x$$

为例，讨论其差分差商在Mizar中是如何实现的。

Mizar 中函数

$$f(x) = \sin^2 x \cos x$$

用 sin(#)sin(#)cos 来表示，此函数的向前差分定理表述如下：

```
theorem :: DIFF_2:62
  for x holds fD(sin(#)sin(#)cos,h).x
  =(1/2)*(sin((6*x+3*h)/2)*sin(3*h/2)-sin((2*x+h)/2)*sin(h/2));
```

根据 MML 中已有的向前差分定义，完成其 Mizar 实现的推理如下：

```
proof
    let x;
    set y=3*x;
```

```
  set z=3*h;
   fD(sin(#)sin(#)cos,h).x
  = (sin(#)sin(#)cos).(x+h)-(sin(#)sin(#)cos).x
::>                                        *4
 .= ((sin(#)sin).(x+h))*(cos.(x+h))-(sin(#)sin(#)cos).x
::>                                              *4
 .= (1/4)*(-cos((x+h)+(x+h)-(x+h))+cos((x+h)+(x+h)-(x+h))
 +cos((x+h)+(x+h)-(x+h))-cos((x+h)+(x+h)+(x+h)))-sin(x)*sin(x)*cos(x)
::>                                                                  *4
 .= (1/4)*(-2*(sin((2*x+h)/2)*sin(h/2)))
  -(1/4)*(-2*(sin((y+z+y)/2)*sin((y+z-y)/2)));
::>                                      *4
  hence thesis;
::>          *4
end;
```

为了使证明过程简洁，我们令 y=3x，z=3h。而“*4”则表示在 Mizar 系统中运行验证命令 mizf 后，检测有错，结论缺少依据。

经修改完善，最终此定理在Mizar中实现的完整证明过程如下：

```
proof
  let x;
  set y=3*x;
  set z=3*h;
  fD(sin(#)sin(#)cos,h).x = (sin(#)sin(#)cos).(x+h)
  -(sin(#)sin(#)cos).x by DIFF_1:3
    .= ((sin(#)sin).(x+h))*(cos.(x+h))
  -(sin(#)sin(#)cos).x by VALUED_1:5
    .= (sin.(x+h))*(sin.(x+h))*(cos.(x+h))
  -(sin(#)sin(#)cos).x by VALUED_1:5
 .= (sin.(x+h))*(sin.(x+h))*(cos.(x+h))
-((sin(#)sin).x)*(cos.x) by VALUED_1:5
 .= sin(x+h)*sin(x+h)*cos(x+h)
-sin(x)*sin(x)*cos(x) by VALUED_1:5
 .= (1/4)*(-cos((x+h)+(x+h)-(x+h))+cos((x+h)+(x+h)-(x+h))
+cos((x+h)+(x+h)-(x+h))-cos((x+h)+(x+h)+(x+h)))
-sin(x)*sin(x)*cos(x) by SIN_COS4:38
 .= (1/4)*(cos(x+h)-cos(3*(x+h)))-(1/4)*(-cos(x+x-x)
+cos(x+x-x)+cos(x+x-x)-cos(x+x+x)) by SIN_COS4:38
 .= (1/4)*(cos(x+h)-cos(x))-(1/4)*(cos(3*(x+h))-cos(3*x))
```

```
  .= (1/4)*(-2*(sin((x+h+x)/2)*sin((x+h-x)/2)))
 -(1/4)*(cos(3*(x+h))-cos(3*x)) by SIN_COS4:22
  .= (1/4)*(-2*(sin((2*x+h)/2)*sin(h/2)))
 -(1/4)*(-2*(sin((y+z+y)/2)*sin((y+z-y)/2))) by SIN_COS4:22
  .= (1/2)*(sin((6*x+3*h)/2)*sin(3*h/2)) -(1/2)*((sin((2*x+h)/2)*sin(h/2)));
 hence thesis;
end;
```

Mizar中每一步的推导都需要由by（或from）引导的定义、定理或前面推导成立的结论，共同作为命题成立的依据。上述证明过程中DIFF_1:3表示Mizar论文DIFF_1.miz里面的定理3，VALUED_1:5表示Mizar论文VALVED_1.miz里面的定理5，SIN_COS4:38表示Mizar论文SIN_COS4.miz里面的定理38，SIN_COS4:22表示Mizar论文SIN_COS4.miz里面的定理22。

同理，函数

$$f(x)=\sin^2 x\cos x$$

的向后差分定理表述及证明过程如下：

```
theorem :: DIFF_2:63
  for x holds bD(sin(#)sin(#)cos,h).x
  = (1/2)*(sin((6*x-3*h)/2)*sin(3*h/2))-(1/2)*(sin((2*x-h)/2)*sin(h/2));
proof
  let x;
  set y=3*x;
  set z=3*h;
  bD(sin(#)sin(#)cos,h).x = (sin(#)sin(#)cos).x
  -(sin(#)sin(#)cos).(x-h) by DIFF_1:4
    .= ((sin(#)sin).x)*(cos.x)
  -(sin(#)sin(#)cos).(x-h) by VALUED_1:5
    .= (sin.x)*(sin.x)*(cos.x)
  -(sin(#)sin(#)cos).(x-h) by VALUED_1:5
    .= (sin.x)*(sin.x)*(cos.x)
  -((sin(#)sin).(x-h))*(cos.(x-h)) by VALUED_1:5
    .= sin(x)*sin(x)*cos(x)
  -sin(x-h)*sin(x-h)*cos(x-h) by VALUED_1:5
    .= (1/4)*(-cos(x+x-x)+cos(x+x-x)+cos(x+x-x)-cos(x+x+x))
  -sin(x-h)*sin(x-h)*cos(x-h) by SIN_COS4:38
    .= (1/4)*(cos(x)-cos(3*x))-(1/4)*(-cos((x-h)+(x-h)-(x-h))
  +cos((x-h)+(x-h)-(x-h))+cos((x-h)+(x-h)-(x-h))
  -cos((x-h)+(x-h)+(x-h))) by SIN_COS4:38
```

```
    .= (1/4)*(cos(x)-cos(x-h))-(1/4)*(cos(3*x)-cos(3*(x-h)))
    .= (1/4)*(-2*(sin((x+(x-h))/2)*sin((x-(x-h))/2)))
  -(1/4)*(cos(3*x)-cos(3*(x-h))) by SIN_COS4:22
    .= (1/4)*(-2*(sin((2*x-h)/2)*sin(h/2))) -(1/4)*(-2*(sin((y+(y-z))/2)*
  sin((y-(y-z))/2))) by SIN_COS4:22
    .= (1/2)*(sin((6*x-3*h)/2)*sin(3*h/2))
  -(1/2)*(sin((2*x-h)/2)*sin(h/2));
  hence thesis;
end;
```

函数

$$f(x)=\sin^2 x\cos x$$

的中心差分定理表述及证明过程如下：

```
 theorem :: DIFF_2:64
  for x holds cD(sin(#)sin(#)cos,h).x
  = -(1/2)*(sin(x)*sin(h/2))+(1/2)*(sin(3*x)*sin(3*h/2));
 proof
  let x;
  set y=3*x;
  set z=3*h;
  cD(sin(#)sin(#)cos,h).x = (sin(#)sin(#)cos).(x+h/2)
  - (sin(#)sin(#)cos).(x-h/2) by DIFF_1:5
    .= ((sin(#)sin).(x+h/2))*(cos.(x+h/2))
  -(sin(#)sin(#)cos).(x-h/2) by VALUED_1:5
    .= (sin.(x+h/2))*(sin.(x+h/2))*(cos.(x+h/2))
  -(sin(#)sin(#)cos).(x-h/2) by VALUED_1:5
    .= (sin.(x+h/2))*(sin.(x+h/2))*(cos.(x+h/2))
  -((sin(#)sin).(x-h/2))*(cos.(x-h/2)) by VALUED_1:5
    .= sin(x+h/2)*sin(x+h/2)*cos(x+h/2)
  -sin(x-h/2)*sin(x-h/2)*cos(x-h/2) by VALUED_1:5
    .= (1/4)*(-cos((x+h/2)+(x+h/2)-(x+h/2))+cos((x+h/2)+(x+h/2)-(x+h/2))
  +cos((x+h/2)+(x+h/2)-(x+h/2))-cos((x+h/2)+(x+h/2)+(x+h/2)))
  -sin(x-h/2)*sin(x-h/2)*cos(x-h/2) by SIN_COS4:38
    .= (1/4)*(cos(x+h/2)-cos(3*(x+h/2)))
  -(1/4)*(-cos((x-h/2)+(x-h/2)-(x-h/2))+cos((x-h/2)+(x-h/2)
  -(x-h/2))+cos((x-h/2)+(x-h/2)-(x-h/2))
  -cos((x-h/2)+(x-h/2)+(x-h/2))) by SIN_COS4:38
    .= (1/4)*(cos(x+h/2)-cos(x-h/2))-(1/4)*(cos(3*(x+h/2))-cos(3*(x-h/2)))
```

```
  .= (1/4)*(-2*(sin(((x+h/2)+(x-h/2))/2)*sin(((x+h/2)-(x-h/2))/2)))
-(1/4)*(cos(3*(x+h/2))-cos(3*(x-h/2))) by SIN_COS4:22
  .= (1/4)*(-2*(sin(x)*sin(h/2))) -(1/4)*(-2*(sin(((y+z/2)+(y-z/2))/2)
*sin(((y+z/2)-(y-z/2))/2))) by SIN_COS4:22
  .= -(1/2)*(sin(x)*sin(h/2))+(1/2)*(sin(3*x)*sin(3*h/2));
hence thesis;
end;
```

根据 MML 中已有的差商定义，函数

$$f(x) = \sin^2 x \cos x$$

的一阶差商定理表述及证明过程如下：

```
theorem :: DIFF_2:61
  [!sin(#)sin(#)cos,x0,x1!]
 = -(1/2)*(sin(3*(x1+x0)/2)*sin(3*(x1-x0)/2)
  +sin((x0+x1)/2)*sin((x0-x1)/2))/(x0-x1);
proof
  set y = 3*x0;
  set z = 3*x1;
  [!sin(#)sin(#)cos,x0,x1!] = (((sin(#)sin).x0)*(cos.x0)
  -(sin(#)sin(#)cos).x1)/(x0-x1) by VALUED_1:5
    .= ((sin.x0)*(sin.x0)*(cos.x0)
  -(sin(#)sin(#)cos).x1)/(x0-x1) by VALUED_1:5
    .= ((sin.x0)*(sin.x0)*(cos.x0)
  -((sin(#)sin).x1)*(cos.x1))/(x0-x1) by VALUED_1:5
    .= (sin(x0)*sin(x0)*cos(x0)
  -sin(x1)*sin(x1)*cos(x1))/(x0-x1) by VALUED_1:5
 .= ((1/4)*(-cos(x0+x0-x0)+cos(x0+x0-x0) +cos(x0+x0-x0)-cos(x0+x0+x0))
  -sin(x1)*sin(x1)*cos(x1))/(x0-x1) by SIN_COS4:38
    .= ((1/4)*(cos(x0)-cos(3*x0))-(1/4)
  *(-cos(x1+x1-x1)+cos(x1+x1-x1)+cos(x1+x1-x1)
  -cos(x1+x1+x1)))/(x0-x1) by SIN_COS4:38
    .= ((1/4)*(cos(x0)-cos(x1))+(1/4)*(cos(z)-cos(y)))/(x0-x1)
    .= ((1/4)*(-2*(sin((x0+x1)/2)*sin((x0-x1)/2)))
  +(1/4)*(cos(z)-cos(y)))/(x0-x1) by SIN_COS4:22
    .= ((1/4)*(-2*(sin((x0+x1)/2)*sin((x0-x1)/2)))+(1/4)
  *(-2*(sin((z+y)/2)*sin((z-y)/2))))/(x0-x1) by SIN_COS4:22
    .= (-(1/2)*(sin(3*(x1+x0)/2)*sin(3*(x1-x0)/2)
  +sin((x0+x1)/2)*sin((x0-x1)/2)))/(x0-x1);
  hence thesis by XCMPLX_1:188;
```

**end;**

其中 x0，x1 是任意实数，(#)表示函数的乘积。

函数

$$f(x)=\sin^2 x\cos x$$

的向前差分、向后差分、中心差分及一阶差商公式在 Mizar 论文中的表述为[24]：

(62) Let given $x$. Then (fD((the function sin) (the function sin) (the function cos), $h$))($x$) $= \frac{1}{2}\cdot(\sin(\frac{6\cdot x+3\cdot h}{2})\cdot\sin(\frac{3\cdot h}{2})-\sin(\frac{2\cdot x+h}{2})\cdot\sin(\frac{h}{2}))$.

(63) Let given $x$. Then (bD((the function sin) (the function sin) (the function cos), $h$))($x$) $= \frac{1}{2}\cdot(\sin(\frac{6\cdot x-3\cdot h}{2})\cdot\sin(\frac{3\cdot h}{2}))-\frac{1}{2}\cdot(\sin(\frac{2\cdot x-h}{2})\cdot\sin(\frac{h}{2}))$.

(64) For every $x$ holds (cD((the function sin) (the function sin) (the function cos), $h$))($x$) $= -\frac{1}{2}\cdot(\sin x\cdot\sin(\frac{h}{2}))+\frac{1}{2}\cdot(\sin(3\cdot x)\cdot\sin(\frac{3\cdot h}{2}))$.

(61) $\Delta$((the function sin) (the function sin) (the function cos), $x_0, x_1$) $= -\frac{\frac{1}{2}\cdot(\sin(\frac{3\cdot(x_1+x_0)}{2})\cdot\sin(\frac{3\cdot(x_1-x_0)}{2})+\sin(\frac{x_0+x_1}{2})\cdot\sin(\frac{x_0-x_1}{2}))}{x_0-x_1}$.

## 3.3 Mizar实现的最终形式

一篇Mizar文章撰写完成后，经过第二章第六节中所有校验命令的检测后都没有出错的文章就是Mizar系统中的最终表现形式了，DIFF_2.miz经mizf最终检测后，结果见图3-1：




图3-1：DIFF_2.miz的检验结果

Fig.3-1 The result of DIFF_2.miz

mizf在检验逻辑推理过程之前，先检验文章环境部的设置和基本的语法问题，标明出错的个数和位置，便于查找修改。DOS下对DIFF_2.miz的检验结果显示：环境部中包含8部分，全文共1472行，Mizar系统检验总耗时25秒钟。在Parser、Analyzer和Checker三部分均显示没有错误，证明文章语法合理、证明正确。

另外，***Formalized Mathematics***中的文本形式也是特殊函数差分差商的Mizar实现形式之一[24]。

## 3.4 成果出处

本文关于特殊函数差分差商的Mizar实现结果收录在Mizar版本Mizar version: 7.8.06_4.91.994中。

# 4 特殊函数积分的Mizar实现

积分学与微分学联系密切，共同组成了分析学的一个分支——微积分学[25]。积分学主要研究积分的性质与计算，而积分公式在自然科学与技术科学中的应用是基础，起到了重要作用。不定积分是求导的逆运算，定积分则是某种特殊和式的极限。牛顿—莱布尼茨公式不仅为定积分计算提供了一个有效的方法，而且在理论上把定积分与不定积分联系了起来[26]。

定理[25]：若函数 $f$ 在 $[a,b]$ 上连续，且存在原函数 $F$，即 $F'(x)=f(x)$ $x\in[a,b]$，则 $f$ 在 $[a,b]$ 上可积，且

$$\int_a^b f(x)dx = F(b)-F(a)$$

这称为牛顿—莱布尼茨公式。此公式常写成

$$\int_a^b f(x)dx = F(x)\big|_{x=b} - F(x)\big|_{x=a}$$

在Mizar中积分也有着重要的意义。本章通过Mizar中的定义、定理以及相关的微分公式实现积分公式的Mizar论证。

## 4.1 积分公式

我们基于德国数学家Fritz Chemnitius教授编写的《微积分特例》[27]，从书中多个特殊函数及其复合的积分公式，选取了部分具有代表性的函数，在Mizar中论证了它们的积分公式，完成了这些积分公式的Mizar实现。本章只选取部分定理给出证明方法和Mizar表述的说明，所有的证明分为2个部分，收录于MML中的INTEGRA9.miz[28]、INTEGR11.miz[29]两篇文章中，发表于2009年的Mizar期刊 ***Formalized Mathematics***。

## 4.2 函数积分的Mizar实现

正如加法有其逆运算减法，乘法有其逆运算除法一样，微分法也有它的逆运算——积分法。Mizar 中特殊函数的积分建立在原函数微分的基础上，所以对于 Mizar 中积分公式的证明，先运用函数的运算法则和微分法则推导出原函数的微分公式，再利用原函数的微分和牛顿—莱布尼茨公式推导被积函数的积分公式。

### 4.2.1 特殊函数积分的环境部设置与变量声明

Mizar论文撰写完毕后环境部的最终形式如下：

**environ**
**vocabularies** ARYTM, RELAT_1, FUNCT_1, ARYTM_1, JGRAPH_2, SEQ_1, SIN_COS, INTEGRA1, FDIFF_1, BOOLE, SQUARE_1, ARYTM_3, ORDINAL2, RCOMP_1, PREPOWER, LATTICES, SIN_COS2, PRE_TOPC, PARTFUN1, TAYLOR_1, FINSEQ_1, SIN_COS4, SIN_COS6, SIN_COS9, SUBSET_1;
**notations** TARSKI, XBOOLE_0, SEQ_1, SIN_COS, SUBSET_1, NUMBERS, VALUED_1, NAT_1, XXREAL_0, REAL_1, RELAT_1, RELSET_1, PARTFUN1, PARTFUN2, RCOMP_1, RFUNCT_1, PSCOMP_1, INTEGRA1, FCONT_1, SQUARE_1, INTEGRA5, PREPOWER, TAYLOR_1, FDIFF_1, JORDAN16, SEQ_2, SIN_COS2, FDIFF_9, SIN_COS6, SIN_COS9, BINOP_2;
**constructors** SIN_COS, TAYLOR_1, SEQ_1, REAL_1, FDIFF_1, FCONT_1, PSCOMP_1, BINOP_2, SQUARE_1, PREPOWER, INTEGRA5, JORDAN16, SIN_COS2, SEQ_4, SIN_COS9, PARTFUN1, PARTFUN2, FUNCT_1, RCOMP_1, RELAT_1, SIN_COS6, RFUNCT_1, VALUED_1, FDIFF_9, SIN_COS4, LIMFUNC1, NAT_1, POWER;
**registrations** VALUED_1, NAT_1, XBOOLE_0, NUMBERS, MEMBERED, VALUED_0, INTEGRA1, INT_1, RELAT_1, ORDINAL1, FUNCT_2, RCOMP_1, FCONT_1, TOPREALB, RELSET_1, RFUNCT_1, XREAL_0, NEWTON, SQUARE_1, FUNCT_1, XXREAL_0, SEQM_3, SIN_COS9, FINSET_1, TAYLOR_1, FCONT_3;
**requirements** NUMERALS, BOOLE, SUBSET, ARITHM;
**definitions** SIN_COS, VALUED_1, SUBSET_1, SQUARE_1, SIN_COS4, SIN_COS6, TARSKI, FCONT_1, FDIFF_9, XBOOLE_0, SIN_COS9, RCOMP_1, LIMFUNC1, NAT_1;
**theorems** PARTFUN1, RFUNCT_1, FUNCT_1, FDIFF_1, FUNCT_2, SQUARE_1, XCMPLX_1, INTEGRA5, INTEGRA8, SIN_COS, FDIFF_4, TAYLOR_1, VALUED_1, FCONT_1, JORDAN16, SIN_COS5, SIN_COS4, SIN_COS2, SIN_COS6, XXREAL_1, XBOOLE_1, FDIFF_5, FDIFF_7, FDIFF_8, FDIFF_9, SIN_COS9, PREPOWER, FDIFF_6;

文章中需要声明的变量数据类型如下：

**begin**
**reserve** x,a,b for Real;
**reserve** n,m for Element of NAT;
**reserve** A for closed-interval Subset of REAL;
**reserve** Z for open Subset of REAL;

### 4.2.2 特殊函数积分的Mizar表述与实现

完成特殊函数在Mizar中的实现分为两步：一是定义新的算子；二是重新组合已有算子。由于在Mizar中定义任何的新算子（functor）都必须给出存在性与唯一性的证明，为了简化所讨论函数的定义，在函数积分的Mizar实现中，我们采用特殊函数的第二种表现形式。

我们将讨论的特殊函数分为几类：指数函数、三角函数、幂函数、双曲函数、无理函数、反三角函数。对于不同的函数，其构造形式也不同，下面从一般的积分公式和Mizar中的表述两方面分类举例说明。

指数函数类，如

$$\int_A e^{-x}dx = (-e^{-x})\Big|_{x=\sup A} - (-e^{-x})\Big|_{x=\inf A}$$

（A 为积分区间，下同）。在 Mizar 中的表述为：

```
:: f.x = exp_R.(-x)
theorem :: INTEGRA9:2
 integral(exp_R*(AffineMap(-1,0)),A)
= (-(exp_R*(AffineMap(-1,0)))).(sup A)
  -(-(exp_R*(AffineMap(-1,0)))).(inf A);
```

其中 exp_R*(AffineMap(-1,0))里的*表示两个函数的复合，exp_R 和 AffineMap(-1,0) 是 Mizar 中函数 $f(x)=e^x$ 和 $f(x)=-x$ 的表述，A 表示闭区间，integral 表示积分。

三角函数类，如 $n\neq 0$ 时

$$\int_A \cos nx dx = \frac{1}{n}\sin nx\Big|_{x=\sup A} - \frac{1}{n}\sin nx\Big|_{x=\inf A}。$$

在 Mizar 中的表述为：

```
:: f.x=cos(n*x)
theorem :: INTEGRA9:13
  n<>0 implies integral(cos*(AffineMap(n,0)),A)
  = ((1/n)(#)(sin*(AffineMap(n,0)))).(sup A)
    -((1/n)(#)(sin*(AffineMap(n,0)))).(inf A)
```

其中 n 为正整数（在变量数据类型中已设定）。

幂函数类，如

$$\int_A x^n dx = \frac{1}{n+1}x^{n+1}\Big|_{x=\sup A} - \frac{1}{n+1}x^{n+1}\Big|_{x=\inf A}。$$

在 Mizar 中的表述为：

```
:: f.x = x^n
```

```
theorem :: INTEGRA9:25
  integral(( #Z n),A)
 = ((1/(n+1))(#)( #Z (n+1))).(sup A)
  -((1/(n+1))(#)( #Z (n+1))).(inf A)
```

此定理中( #Z n)是函数 $f(x)=x^n$ 的 Mizar 表述，(1/(n+1))(#)( #Z (n+1))里的（#）表示乘号。

双曲函数类，如

$$\int_A x\sinh x dx = (x\cosh x - \sinh x)\Big|_{x=\sup A} - (x\cosh x - \sinh x)\Big|_{x=\inf A}。$$

在 Mizar 中的表述为：

```
::  f.x = x*sinh.x
theorem :: INTEGR11:30
 integral((AffineMap(1,0))(#)sinh,A)
 = ((AffineMap(1,0))(#)cosh-sinh).(sup A)
  -((AffineMap(1,0))(#)cosh-sinh).(inf A)
```

无理函数类，如

$$\int_A \sqrt{a+x}dx = \frac{2}{3}(a+x)^{\frac{3}{2}}\Big|_{x=\sup A} - \frac{2}{3}(a+x)^{\frac{3}{2}}\Big|_{x=\inf A}。$$

在 Mizar 中的表述为：

```
reserve Z for open Subset of REAL;
:: f.x = (a+x) #R (1/2)
theorem :: INTEGRA9:70
  A c= Z & (for x st x in Z holds f.x=a+x & f.x>0)
  & dom ((2/3)(#)(( #R (3/2))*f)) = Z
  & dom ((2/3)(#)(( #R (3/2))*f)) = dom f2
  & (for x st x in Z holds f2.x = (a+x) #R (1/2)) &
  f2|A is continuous implies
  integral(f2,A) = ((2/3)(#)(( #R (3/2))*f)).(sup A)
           -((2/3)(#)(( #R (3/2))*f)).(inf A);
```

这里首先在论文开始变量数据类型中设定 Z 为开区间，( #R (3/2))*f 表示 $f^{\frac{3}{2}}$，f2|A is continuous 表示函数 f2 在闭区间 A 上连续。

反三角函数类，如

$$\int_A \arcsin x dx = \left(x\arcsin x + \sqrt{1-x^2}\right)\Big|_{x=\sup A} - \left(x\arcsin x + \sqrt{1-x^2}\right)\Big|_{x=\inf A}。$$

在 Mizar 中的表述为：

```
:: f.x = arcsin.x
theorem :: INTEGR11:68
  A c= Z & Z c= ]. -1,1 .[ & f=f1-f2 & f2=#Z 2
  & (for x st x in Z holds f1.x=1 & f.x >0 & x<>0) & dom arcsin=Z
  & Z c= dom ((id Z)(#)(arcsin)+( #R (1/2))*f) implies
  integral(arcsin,A) = ((id Z)(#)(arcsin)+( #R (1/2))*f).(sup A)
     -((id Z)(#)(arcsin) + ( #R (1/2))*f).(inf A);
```

其中]. -1,1 .[ 表示-1到1的开区间，id Z 在Mizar中表述如下：

```
definition
  let Z;
  func   id Z -> Relation means
  [x,y] in it iff x in Z & x = y;
end;
```

即 $\forall x \in Z$，函数 $f(x)=x$。

以上特殊函数的积分公式及推证过程参见Mizar文章INTEGRA9和INTEGR11，其在Mizar论文中的表述分别为[28,29]：

函数 $f(x)=e^{-x}$：

(2) $\int_A ((\text{the function exp}) \cdot \text{AffineMap}(-1, 0))(x)dx = (-(\text{the function exp}) \cdot \text{AffineMap}(-1,$ $(-(\text{the function exp}) \cdot \text{AffineMap}(-1, 0))(\inf A)$.

函数 $f(x)=\cos nx\ (n \neq 0)$：

(13) If $n \neq 0$, then $\int_A ((\text{the function cos}) \cdot \text{AffineMap}(n, 0))(x)dx =$ $(\frac{1}{n}((\text{the function sin}) \cdot \text{AffineMap}(n, 0)))(\sup A) - (\frac{1}{n}((\text{the function sin})$ $\cdot \text{AffineMap}(n, 0)))(\inf A)$.

函数 $f(x)=x^n$：

(25) $\int_A (\frac{n}{\mathbb{Z}})(x)dx = (\frac{1}{n+1}\frac{n+1}{\mathbb{Z}})(\sup A) - (\frac{1}{n+1}\frac{n+1}{\mathbb{Z}})(\inf A)$.

函数 $f(x)=x\sinh x$：

(30) $\int_A$(AffineMap(1,0) (the function sinh))$(x)dx$ = (AffineMap(1,0) (the function cosh)−the function sinh)(sup $A$) − (AffineMap(1,0) (the function cosh)−the function sinh)(inf $A$).

函数 $f(x)=\sqrt{a+x}$：

(70) Suppose that
(i) $A \subseteq Z$,
(ii) for every $x$ such that $x \in Z$ holds $f(x) = a + x$ and $f(x) > 0$,
(iii) $\mathrm{dom}(\frac{2}{3}\,((\frac{3}{2})_{\mathbb{R}} \cdot f)) = Z$,
(iv) $\mathrm{dom}(\frac{2}{3}\,((\frac{3}{2})_{\mathbb{R}} \cdot f)) = \mathrm{dom}\, f_2$,
(v) for every $x$ such that $x \in Z$ holds $f_2(x) = (a + x)^{\frac{1}{2}}_{\mathbb{R}}$, and
(vi) $f_2 \restriction A$ is continuous.

Then $\int_A f_2(x)dx = (\frac{2}{3}\,((\frac{3}{2})_{\mathbb{R}} \cdot f))(\sup A) - (\frac{2}{3}\,((\frac{3}{2})_{\mathbb{R}} \cdot f))(\inf A)$.

函数 $f(x)=\arcsin x$：

(68) Suppose that $A \subseteq Z$ and $Z \subseteq ]-1,1[$ and $f = f_1 - f_2$ and $f_2 = \frac{2}{Z}$ and for every $x$ such that $x \in Z$ holds $f_1(x) = 1$ and $f(x) > 0$ and $x \neq 0$ and dom (the function arcsin) $= Z$ and $Z \subseteq$ dom($\mathrm{id}_Z$ (the function arcsin)+$(\frac{1}{2})_{\mathbb{R}} \cdot f$). Then $\int_A$(the function arcsin)$(x)dx$ = ($\mathrm{id}_Z$ (the function arcsin)+$(\frac{1}{2})_{\mathbb{R}} \cdot f$)(sup $A$) − ($\mathrm{id}_Z$ (the function arcsin)+$(\frac{1}{2})_{\mathbb{R}} \cdot f$)(inf $A$).

### 4.2.3 特殊函数积分的Mizar证明

首先研究三角函数

$$f(x)=\sin mx\cos nx(m\neq n),$$

讨论其积分公式在Mizar中是如何实现的。

一般的数学公式为，$m\neq n$ 时

$$\int_A \sin mx\cos nx dx = \left(-\frac{\cos(m+n)x}{2(m+n)}-\frac{\cos(m-n)x}{2(m-n)}\right)\Big|_{x=\sup A}-\left(-\frac{\cos(m+n)x}{2(m+n)}-\frac{\cos(m-n)x}{2(m-n)}\right)\Big|_{x=\inf A}$$

在 Mizar 中定理表述如下：

```
reserve n,m for Element of NAT;
reserve A for closed-interval Subset of REAL;
 :: f.x = sin.(m*x)*cos.(n*x)
 theorem :: INTEGR11:27
  m+n<>0 & m-n<>0 implies
```

```
integral((sin*AffineMap(m,0))(#)(cos*AffineMap(n,0)),A)
= ((-(1/(2*(m+n)))(#)(cos*AffineMap(m+n,0)))-
(1/(2*(m-n)))(#)(cos*AffineMap(m-n,0))).(sup A)
 -((-(1/(2*(m+n)))(#)(cos*AffineMap(m+n,0)))-
(1/(2*(m-n)))(#)(cos*AffineMap(m-n,0))).(inf A);
```

证明之前设定$n$，$m$为自然数，A 是闭区间。根据原函数的可微性与微分公式来求导被积函数的积分公式。由$m+n\neq 0$和$m-n\neq 0$可知积分后所得到的函数有意义，并说明$m\neq n$。此定理在 Mizar 中实现的完整证明过程如下（下划线标注的是主要推理思路与步骤）：

```
proof
 assume A: m+n<>0 & m-n<>0;
A1: [#]REAL=dom ((sin*AffineMap(m,0))(#)(cos*AffineMap(n,0)))
  & dom (sin*AffineMap(m,0)) = [#]REAL
  & dom (cos*AffineMap(n,0)) = [#]REAL by FUNCT_2:def 1;
then C: dom (sin*AffineMap(m,0)) /\ dom (cos*AffineMap(n,0)) = [#]REAL;
A2: dom (AffineMap(m,0))=[#]REAL & dom (AffineMap(n,0))=[#]REAL
by FUNCT_2:def 1;
  for x st x in REAL holds
 AffineMap(n,0).x=n*x + 0 by JORDAN16:def 3; then
  AffineMap(n,0)|REAL is continuous
  by A2,FDIFF_1:31,FDIFF_1:33;then
A3: AffineMap(n,0)|A is continuous by FCONT_1:17;
 cos|((AffineMap(n,0)).:A) is continuous;then
A5: (cos*AffineMap(n,0))|A is continuous by A3,FCONT_1:26;
  for x st x in REAL holds
    AffineMap(m,0).x=m*x + 0 by JORDAN16:def 3; then
  AffineMap(m,0)|REAL is continuous
  by A2,FDIFF_1:31,FDIFF_1:33;then
A6: AffineMap(m,0)|A is continuous by FCONT_1:17;
 sin|((AffineMap(m,0)).:A) is continuous;then
 (sin*AffineMap(m,0))|A is continuous by A6,FCONT_1:26;then
 ((sin*AffineMap(m,0))(#)(cos*AffineMap(n,0)))|A
   is continuous by C,A5,FCONT_1:19;then
A7: ((sin*AffineMap(m,0))(#)(cos*AffineMap(n,0))) is integrable on A
 & ((sin*AffineMap(m,0))(#)(cos*AffineMap(n,0)))|A
     is bounded by A1,INTEGRA5:10,11;
A8: ((-(1/(2*(m+n)))(#)(cos*AffineMap(m+n,0)))-
   (1/(2*(m-n)))(#)(cos*AffineMap(m-n,0)))
```

```
    is_differentiable_on REAL by A,Th7;
B1: for x st x in REAL holds AffineMap(m,0).x=m*x
  proof
    let x;
   assume x in REAL;
    (AffineMap(m,0)).x = m*x + 0 by JORDAN16:def 3
      .=m*x;
    hence thesis;
  end;
B2: for x st x in REAL holds AffineMap(n,0).x=n*x
  proof
    let x;
   assume x in REAL;
    (AffineMap(n,0)).x = n*x + 0 by JORDAN16:def 3
      .=n*x;
    hence thesis;
  end;
A9: for x st x in dom (((-(1/(2*(m+n))))(#)(cos*AffineMap(m+n,0)))-
  (1/(2*(m-n)))(#)(cos*AffineMap(m-n,0)))`|REAL) holds
   (((-(1/(2*(m+n))))(#)(cos*AffineMap(m+n,0)))-
  (1/(2*(m-n)))(#)(cos*AffineMap(m-n,0)))`|REAL).x
   = ((sin*AffineMap(m,0))(#)(cos*AffineMap(n,0))).x
  proof
     let x;
     assume x in dom (((-(1/(2*(m+n))))(#)(cos*AffineMap(m+n,0)))-
  (1/(2*(m-n)))(#)(cos*AffineMap(m-n,0)))`|REAL);
      (((-(1/(2*(m+n))))(#)(cos*AffineMap(m+n,0)))-
  (1/(2*(m-n)))(#)(cos*AffineMap(m-n,0)))`|REAL).x
   = sin.(m*x)*cos.(n*x) by A,Th7
  .= sin.(AffineMap(m,0).x)*cos.(n*x) by B1
  .= (sin.(AffineMap(m,0).x))*(cos.(AffineMap(n,0).x)) by B2
  .= ((sin*AffineMap(m,0)).x)*(cos.(AffineMap(n,0).x)) by A1,FUNCT_1:22
  .= ((sin*AffineMap(m,0)).x)*((cos*AffineMap(n,0)).x) by A1,FUNCT_1:22
  .= ((sin*AffineMap(m,0))(#)(cos*AffineMap(n,0))).x by VALUED_1:5;
    hence thesis;
  end;
   dom (((-(1/(2*(m+n))))(#)(cos*AffineMap(m+n,0)))-
   (1/(2*(m-n)))(#)(cos*AffineMap(m-n,0)))`|REAL)
```

```
  = dom ((sin*AffineMap(m,0))(#)(cos*AffineMap(n,0)))
        by A1,A8,FDIFF_1:def 8;then
   (((-(1/(2*(m+n)))(#)(cos*AffineMap(m+n,0)))-
   (1/(2*(m-n)))(#)(cos*AffineMap(m-n,0)))`|REAL)
  = (sin*AffineMap(m,0))(#)(cos*AffineMap(n,0)) by A9,PARTFUN1:34;
   hence thesis by A,A7,Th7,INTEGRA5:13;
end;
```

其中，"proof" 和 "end" 成对出现，缺一不可。若证明过程中后面会引用已知条件和前面已推导成立的结论作为条件，则需在已知条件和结论前加上标号，如：A1，A2，B1，B2，…，C 等。由 by 引导的定义、定理或前面推导成立的结论，除了定理 7(Th7)之外都出自上述证明本身或 MML 中已收录的 Mizar 文章。定理 7（Th7）是为了简化上述定理的证明过程而提前证明的一个小定理，其具体表述为：

定理7（Th7）：$m+n\neq 0$，$m-n\neq 0$ 时，函数 $f(x)=-\frac{\cos(m+n)x}{2(m+n)}-\frac{\cos(m-n)x}{2(m-n)}$ 在定义域——全体实数上可微，且 $\forall x\in R$，$\frac{df}{dx}=\sin mx\cos nx$。

显然，这是一个求函数微分公式的定理。在Mizar中此定理以如下形式表述：

```
:: f.x=-cos.((m+n)*x)/(2*(m+n))-cos.((m-n)*x)/(2*(m-n))
theorem :: INTEGR11:7
  m+n<>0 & m-n<>0 implies ((-(1/(2*(m+n)))(#)(cos*AffineMap(m+n,0))) -
  (1/(2*(m-n)))(#)(cos*AffineMap(m-n,0))) is_differentiable_on REAL
  & for x holds ((-((1/(2*(m+n)))(#)(cos*AffineMap(m+n,0)))-
  (1/(2*(m-n)))(#)(cos*AffineMap(m-n,0)))`|REAL).x
    = sin.(m*x) * cos.(n*x);
```

证明过程参见 INTEGR11.miz[29]。

通过对定理7（Th7）及其它已证结论、定义和定理的引用，分步完成了Mizar系统中特殊函数 $f(x)=\sin mx\cos nx(m\neq n)$ 的积分公式的推理论证。

类似的，三角函数 $f(x)=\cos mx\cos nx(m\neq n)$，$f(x)=\sin mx\sin nx(m\neq n)$，$f(x)=x\sin nx$，$f(x)=x\cos nx$，$f(x)=\sin^2 x$，$f(x)=\cos^2 x$，$f(x)=\sin^n x\cos x$，$f(x)=\cos^n x\sin x$ 的积分公式在Mizar系统中表述如下（证明的详细过程参 INTEGR11.miz[29]）：

```
   :: f.x = cos.(m*x)*cos.(n*x)
   theorem
     m+n<>0 & m-n<>0 implies
    integral((cos*AffineMap(m,0))(#)(cos*AffineMap(n,0)),A)
```

```
 = ((1/(2*(m+n)))(#)(sin*AffineMap(m+n,0))+
 (1/(2*(m-n)))(#)(sin*AffineMap(m-n,0))).(sup A)
  -((1/(2*(m+n)))(#)(sin*AffineMap(m+n,0))+
 (1/(2*(m-n)))(#)(sin*AffineMap(m-n,0))).(inf A);

:: f.x = sin.(m*x)*sin.(n*x)
theorem
 m+n<>0 & m-n<>0 implies
 integral((sin*AffineMap(m,0))(#)(sin*AffineMap(n,0)),A)
 = ((1/(2*(m-n)))(#)(sin*AffineMap(m-n,0))-
 (1/(2*(m+n)))(#)(sin*AffineMap(m+n,0))).(sup A)
  -((1/(2*(m-n)))(#)(sin*AffineMap(m-n,0))-
 (1/(2*(m+n)))(#)(sin*AffineMap(m+n,0))).(inf A);

:: f.x = x*sin.(n*x)
theorem
 n<>0 implies integral((AffineMap(1,0))(#)(sin*AffineMap(n,0)),A)
 = ((1/(n^2))(#)(sin*AffineMap(n,0))-
 (AffineMap(1/n,0))(#)(cos*AffineMap(n,0))).(sup A)
  -((1/(n^2))(#)(sin*AffineMap(n,0))-
 (AffineMap(1/n,0))(#)(cos*AffineMap(n,0))).(inf A);

:: f.x = x*cos.(n*x)
theorem
 n<>0 implies integral((AffineMap(1,0))(#)(cos*AffineMap(n,0)),A)
 = ((1/(n^2))(#)(cos*AffineMap(n,0))+
 (AffineMap(1/n,0))(#)(sin*AffineMap(n,0))).(sup A)
  -((1/(n^2))(#)(cos*AffineMap(n,0))+
 (AffineMap(1/n,0))(#)(sin*AffineMap(n,0))).(inf A);

:: f.x = (sin.x)^2
theorem
  integral(sin^2,A)
 = (AffineMap(1/2,0)-(1/4)(#)(sin*AffineMap(2,0))).(sup A)
  -(AffineMap(1/2,0)-(1/4)(#)(sin*AffineMap(2,0))).(inf A);

:: f.x = (cos.x)^2
theorem
```

```
  integral(cos^2,A)
    = (AffineMap(1/2,0)+(1/4)(#)(sin*AffineMap(2,0))).(sup A)
    -(AffineMap(1/2,0)+(1/4)(#)(sin*AffineMap(2,0))).(inf A);

:: f.x = (sin.x)^n*(cos.x)
theorem
  integral(( #Z n*sin)(#)cos,A)
 = ((1/(n+1))(#)( #Z (n+1)*sin)).(sup A)
  -((1/(n+1))(#)( #Z (n+1)*sin)).(inf A);

:: f.x = (cos.x)^n*(sin.x)
theorem
  integral(( #Z n*cos)(#)sin,A)
 = ((-1/(n+1))(#)( #Z (n+1)*cos)).(sup A)
  -((-1/(n+1))(#)( #Z (n+1)*cos)).(inf A);
```

接下来研究函数 $f(x)=(ax+b)^n$ 积分的 Mizar 实现过程。我们知道

$$\int_A (ax+b)^n dx=(\frac{1}{a(n+1)}(ax+b)^{n+1})\Big|_{x=\sup A}-(\frac{1}{a(n+1)}(ax+b)^{n+1})\Big|_{x=\inf A},$$

在 Mizar 中表述为：

```
:: f.x = (a*x+b)^n
theorem
 a*(n+1) <> 0 implies integral( #Z n*AffineMap(a,b),A)
 = ((1/(a*(n+1)))(#)( #Z (n+1)*AffineMap(a,b))).(sup A)
  -((1/(a*(n+1)))(#)( #Z (n+1)*AffineMap(a,b))).(inf A);
```

在 Mizar 中推证过程如下（下划线标注的是主要推理思路与步骤）：

```
proof
  assume C: a*(n+1) <> 0;
A1: [#]REAL=dom ( #Z n*AffineMap(a,b))
    & [#]REAL=dom((1/(a*(n+1)))(#)( #Z (n+1)*AffineMap(a,b))) &
    [#]REAL=dom (AffineMap(a,b)) by FUNCT_2:def 1;
    for x st x in REAL holds
      AffineMap(a,b).x=a*x + b by JORDAN16:def 3;then
B1: AffineMap(a,b) is_differentiable_on REAL & for x st x in [#]REAL
      holds ((AffineMap(a,b))`|REAL).x = a by A1,FDIFF_1:31;
A:  #Z n*AffineMap(a,b) is_differentiable_in x
    proof
```

```
      AffineMap(a,b) is_differentiable_in x by B1,FDIFF_1:16;
      hence thesis by TAYLOR_1:3;
    end;
A2: #Z n*AffineMap(a,b) is_differentiable_on REAL
    proof
      for x st x in REAL holds
      #Z n*AffineMap(a,b) is_differentiable_in x by A;
      hence thesis by A1,FDIFF_1:16;
    end;
    ( #Z n*AffineMap(a,b))|REAL is continuous by A2,FDIFF_1:33;then
    ( #Z n*AffineMap(a,b))|A is continuous by FCONT_1:17; then
A3: #Z n*AffineMap(a,b) is_integrable_on A
    & ( #Z n*AffineMap(a,b))|A is bounded by A1,INTEGRA5:10,11;
A4: for x st x in dom (((1/(a*(n+1)))
      (#)( #Z (n+1)*AffineMap(a,b)))`|REAL) holds
      (((1/(a*(n+1)))(#)( #Z (n+1)*AffineMap(a,b)))`|REAL).x
      = ( #Z n*AffineMap(a,b)).x
   proof
      let x;
      assume x in dom (((1/(a*(n+1)))(#)( #Z (n+1)*AffineMap(a,b)))`|REAL);
      (((1/(a*(n+1)))(#)( #Z (n+1)*AffineMap(a,b)))`|REAL).x
        =(a*x+b) #Z n by C,Th12
       .= (AffineMap(a,b).x) #Z n by JORDAN16:def 3
       .= ( #Z n).(AffineMap(a,b).x) by TAYLOR_1:def 1
       .=( #Z n*AffineMap(a,b)).x by A1,FUNCT_1:23;
      hence thesis;
   end;
    (1/(a*(n+1)))(#)( #Z (n+1)*AffineMap(a,b))
       is_differentiable_on REAL by C,Th12; then
    dom (((1/(a*(n+1)))(#)( #Z (n+1)*AffineMap(a,b)))`|REAL)
      = dom ( #Z n*AffineMap(a,b)) by A1,FDIFF_1:def 8;then
    (((1/(a*(n+1)))(#)( #Z (n+1)*AffineMap(a,b)))`|REAL)
      = #Z n*AffineMap(a,b) by A4,PARTFUN1:34;
    hence thesis by C,A3,Th12,INTEGRA5:13;
end;
```

以上特殊函数积分公式在 Mizar 论文中的表述为[29]：

(13) $\int_A$ (the function sin)$^2(x)dx$ = (AffineMap($\frac{1}{2}$,0) $-\frac{1}{4}$ ((the function sin) $\cdot$ AffineMap(2,0)))(sup $A$) $-$ (AffineMap($\frac{1}{2}$,0) $-\frac{1}{4}$ ((the function sin) $\cdot$ AffineMap(2,0)))(inf $A$).

(16) $\int_A$ (the function cos)$^2(x)dx$ = (AffineMap($\frac{1}{2}$,0) $+\frac{1}{4}$ ((the function sin) $\cdot$ AffineMap(2,0)))(sup $A$) $-$ (AffineMap($\frac{1}{2}$,0) $+\frac{1}{4}$ ((the function sin) $\cdot$ AffineMap(2,0)))(inf $A$).

(19) $\int_A$ ((($\frac{n}{\mathbb{Z}}$) $\cdot$ (the function sin)) (the function cos))$(x)dx$ = ($\frac{1}{n+1}$ (($\frac{n+1}{\mathbb{Z}}$) $\cdot$ (the function sin)))(sup $A$) $-$ ($\frac{1}{n+1}$ (($\frac{n+1}{\mathbb{Z}}$) $\cdot$ (the function sin)))(inf $A$).

(22) $\int_A$ ((($\frac{n}{\mathbb{Z}}$) $\cdot$ (the function cos)) (the function sin))$(x)dx$ = (($-\frac{1}{n+1}$) (($\frac{n+1}{\mathbb{Z}}$) $\cdot$ (the function cos)))(sup $A$) $-$ (($-\frac{1}{n+1}$) (($\frac{n+1}{\mathbb{Z}}$) $\cdot$ (the function cos)))(inf $A$).

(25) Suppose $m+n \neq 0$ and $m-n \neq 0$. Then $\int_A$ (((the function cos) $\cdot$ AffineMap($m$,0)) ((the fu

($\frac{1}{2\cdot(m+n)}$ ((the function sin) $\cdot$ AffineMap($m+n$,0)) $+\frac{1}{2\cdot(m-n)}$ ((the function sin) $\cdot$ AffineMap($m-n$,0)))(sup $A$) $-$ ($\frac{1}{2\cdot(m+n)}$ ((the function sin) $\cdot$ AffineMap($m+n$,0)) $+\frac{1}{2\cdot(m-n)}$ ((the function sin) $\cdot$ AffineMap($m-n$,0)))(inf $A$).

(26) Suppose $m+n \neq 0$ and $m-n \neq 0$. Then $\int_A$ (((the function sin) $\cdot$ AffineMap($m$,0)) ((the fu

($\frac{1}{2\cdot(m-n)}$ ((the function sin) $\cdot$ AffineMap($m-n$,0)) $-\frac{1}{2\cdot(m+n)}$ ((the function sin) $\cdot$ AffineMap($m+n$,0)))(sup $A$) $-$ ($\frac{1}{2\cdot(m-n)}$ ((the function sin) $\cdot$ AffineMap($m-n$,0)) $-\frac{1}{2\cdot(m+n)}$ ((the function sin) $\cdot$ AffineMap($m+n$,0)))(inf $A$).

(27) Suppose $m+n \neq 0$ and $m-n \neq 0$. Then $\int_A$ (((the function sin) $\cdot$ AffineMap($m$,0)) ((the fu

($-\frac{1}{2\cdot(m+n)}$ ((the function cos) $\cdot$ AffineMap($m+n$,0)) $-\frac{1}{2\cdot(m-n)}$ ((the function cos) $\cdot$ AffineMap($m-n$,0)))(sup $A$) $-$ ($-\frac{1}{2\cdot(m+n)}$ ((the function cos) $\cdot$ AffineMap($m$

$\frac{1}{2\cdot(m-n)}$ ((the function cos) $\cdot$ AffineMap($m-n$,0)))(inf $A$).

(28) Suppose $n \neq 0$. Then $\int_A$ (AffineMap(1,0) ((the function sin) $\cdot$ AffineMap($n$,0)))$(x)dx$ =

($\frac{1}{n^2}$ ((the function sin) $\cdot$ AffineMap($n$,0)) $-$ AffineMap($\frac{1}{n}$,0) ((the function cos) $\cdot$ AffineMap($n$,0)))(sup $A$) $-$ ($\frac{1}{n^2}$ ((the function sin) $\cdot$ AffineMap($n$,0)) $-$ AffineMap($\frac{1}{n}$,0) ((the function cos) $\cdot$ AffineMap($n$,0)))(inf $A$).

(29) Suppose $n \neq 0$. Then $\int\limits_A$ (AffineMap(1, 0) ((the function cos) $\cdot$ AffineMap($n$, 0)))($x$)$dx$
$= (\frac{1}{n^2}$ ((the function cos) $\cdot$ AffineMap($n$, 0)) + AffineMap($\frac{1}{n}$, 0) ((the function sin) $\cdot$ AffineMap($n$, 0)))(sup $A$)$-$($\frac{1}{n^2}$ ((the function cos) $\cdot$ AffineMap($n$, 0))+ AffineMap($\frac{1}{n}$, 0) ((the function sin) $\cdot$ AffineMap($n$, 0)))(inf $A$).

(32) If $a\cdot(n+1) \neq 0$, then $\int\limits_A$ (($\frac{n}{Z}$) $\cdot$ AffineMap($a$, $b$))($x$)$dx$ = ($\frac{1}{a\cdot(n+1)}$ (($\frac{n+1}{Z}$) $\cdot$ AffineMap($a$, $b$)))(sup $A$) $-$ ($\frac{1}{a\cdot(n+1)}$ (($\frac{n+1}{Z}$) $\cdot$ AffineMap($a$, $b$)))(inf $A$).

再具体一点，给定闭区间 $A=[0,\pi]$，在Mizar系统中实现函数

$$f(x)=\sin 2x\cos x$$

的积分公式。定理表述如下：

```
theorem
  A=[.0,PI.] implies
  integral((sin*AffineMap(2,0))(#)(cos*AffineMap(1,0)),A)=4/3
```

以下是此定理完整的证明过程：

```
proof
  assume A=[.0,PI.];
  then B: sup A=PI & inf A=0 by INTEGRA8:37;
 D:dom (AffineMap(3,0)) = REAL
 & dom (AffineMap(1,0)) = REAL by FUNCT_2:def 1;
  integral((sin*AffineMap(2,0))(#)(cos*AffineMap(1,0)),A)
= ((-(1/(2*(2+1)))(#)(cos*AffineMap(2+1,0)))-
  (1/(2*(2-1)))(#)(cos*AffineMap(2-1,0))).PI
   -((-(1/(2*(2+1)))(#)(cos*AffineMap(2+1,0)))-
  (1/(2*(2-1)))(#)(cos*AffineMap(2-1,0))).0 by B,INTEGR11:27
.= (-(1/6)(#)(cos*AffineMap(3,0))).PI
    -((1/2)(#)(cos*AffineMap(1,0))).PI-((-(1/6)(#)(cos*AffineMap(3,0)))
   -(1/2)(#)(cos*AffineMap(1,0))).0 by VALUED_1:15
.= (-(1/6)(#)(cos*AffineMap(3,0))).PI
   -((1/2)(#)(cos*AffineMap(1,0))).PI
    -((-(1/6)(#)(cos*AffineMap(3,0))).0
   -((1/2)(#)(cos*AffineMap(1,0))).0) by VALUED_1:15
.= (-((1/6)(#)(cos*AffineMap(3,0))).PI)
   -((1/2)(#)(cos*AffineMap(1,0))).PI
    -((-(1/6)(#)(cos*AffineMap(3,0))).0
   -((1/2)(#)(cos*AffineMap(1,0))).0) by VALUED_1:8
.= -(1/6)*((cos*AffineMap(3,0)).PI)
```

```
    -((1/2)(#)(cos*AffineMap(1,0))).PI
     -((-(1/6)(#)(cos*AffineMap(3,0))).0
    -((1/2)(#)(cos*AffineMap(1,0))).0) by VALUED_1:6
.= -(1/6)*((cos*AffineMap(3,0)).PI)
    -(1/2)*((cos*AffineMap(1,0)).PI)
     -((-(1/6)(#)(cos*AffineMap(3,0))).0
    -((1/2)(#)(cos*AffineMap(1,0))).0) by VALUED_1:6
.= -(1/6)*((cos*AffineMap(3,0)).PI)
    -(1/2)*((cos*AffineMap(1,0)).PI)
     -(-(((1/6)(#)(cos*AffineMap(3,0))).0)
    -((1/2)(#)(cos*AffineMap(1,0))).0) by VALUED_1:8
.= -(1/6)*((cos*AffineMap(3,0)).PI)
    -(1/2)*((cos*AffineMap(1,0)).PI)
     -(-(1/6)*((cos*AffineMap(3,0)).0)
    -((1/2)(#)(cos*AffineMap(1,0))).0) by VALUED_1:6
.= -(1/6)*((cos*AffineMap(3,0)).PI)
    -(1/2)*((cos*AffineMap(1,0)).PI)
     -(-(1/6)*((cos*AffineMap(3,0)).0)
    -(1/2)*((cos*AffineMap(1,0)).0)) by VALUED_1:6
.= -(1/6)*(cos.(AffineMap(3,0).PI))
    -(1/2)*((cos*AffineMap(1,0)).PI)
     -(-(1/6)*((cos*AffineMap(3,0)).0)
    -(1/2)*((cos*AffineMap(1,0)).0)) by D,FUNCT_1:23
.= -(1/6)*(cos.(AffineMap(3,0).PI))
    -(1/2)*(cos.(AffineMap(1,0).PI))
     -(-(1/6)*((cos*AffineMap(3,0)).0)
    -(1/2)*((cos*AffineMap(1,0)).0)) by D,FUNCT_1:23
.= -(1/6)*(cos.(AffineMap(3,0).PI))
    -(1/2)*(cos.(AffineMap(1,0).PI))
     -(-(1/6)*(cos.(AffineMap(3,0).0))
    -(1/2)*((cos*AffineMap(1,0)).0)) by D,FUNCT_1:23
.= -(1/6)*(cos.(AffineMap(3,0).PI))
    -(1/2)*(cos.(AffineMap(1,0).PI))
     -(-(1/6)*(cos.(AffineMap(3,0).0))
    -(1/2)*(cos.(AffineMap(1,0).0))) by D,FUNCT_1:23
.= -(1/6)*(cos.(AffineMap(3,0).PI))
    -(1/2)*(cos.(1*PI+0))
     -(-(1/6)*(cos.(AffineMap(3,0).0))
```

```
      -(1/2)*(cos.(AffineMap(1,0).0))) by JORDAN16:def 3
 .= -(1/6)*(cos.(3*PI+0))-(1/2)*cos.PI
       -(-(1/6)*(cos.(AffineMap(3,0).0))
      -(1/2)*(cos.(AffineMap(1,0).0))) by JORDAN16:def 3
 .= -(1/6)*(cos.(2*PI+PI))-(1/2)*cos.PI
       -(-(1/6)*(cos.(3*0+0))
      -(1/2)*(cos.(AffineMap(1,0).0))) by JORDAN16:def 3
 .= -(1/6)*(cos.(2*PI+PI))-(1/2)*cos.PI
       -(-(1/6)*cos.0-(1/2)*cos.(1*0+0)) by JORDAN16:def 3
 .= -(1/6)*(cos.PI)-(1/2)*cos.PI
       -(-(1/6)*cos.0-(1/2)*cos.(1*0+0)) by SIN_COS:83
 .= 4/3 by SIN_COS:33,SIN_COS:81;
    hence thesis;
 end;
```

## 4.3 Mizar 实现的最终形式

关于特殊函数积分的 Mizar 论文 INTEGR11.miz，经 mizf 及其它 Mizar 系统优化命令的校验后的结果见图 4-1：




图 4-1：INTEGR11.miz 的检验结果

Fig.4-1 the result of INTEGR11.miz

Mizar 系统验证结果显示：环境部共包括 8 个部分，Parser、Analyzer、Checker 三部分均无错误显示，全文共 3201 行，验证过程总耗时 41 秒。

另外，特殊函数积分 Mizar 实现的结果[28,29]收录于 2009 年的 ***Formalized Mathematics***。

## 4.4 成果出处

本文关于特殊函数积分的相关结果收录在 Mizar 版本 Mizar version: 7.10.01_4.112.1041 中。

# 5 函数正交性与函数范数的Mizar实现与论证

Mizar中已涉及到积分的定义和其相关的基本性质，但没有做进一步的研究讨论，而函数正交性与函数范数就是建立在积分的基础上，其基本性质在数值逼近中也起着非常重要的作用，所以在Mizar中探索函数正交性和函数范数是十分有意义的。本章的目的在于实现Mizar中函数正交性与函数范数的定义，并用Mizar系统对其相关性质予以论证。

## 5.1 函数正交性与函数范数

### 5.1.1 函数正交性及性质

定义 5.1[30] 函数 $f(x)$ 和 $g(x)$ 的内积表示为

$$(f(x),g(x))=\int_a^b f(x)g(x)dx.$$

内积有以下性质[30]：

1 $(f(x),g(x))=(g(x),f(x))$

2 $\forall r\in R$，有 $(r\times f(x),g(x))=r\times(f(x),g(x))$

3 $(f_1(x)+f_2(x),g(x))=(f_1(x),g(x))+(f_2(x),g(x))$

4 $(f_1(x)-f_2(x),g(x))=(f_1(x),g(x))-(f_2(x),g(x))$

5 $(f(x)+g(x),f(x)+g(x))=(f(x),f(x))+2\times(f(x),g(x))+(g(x),g(x))$

定义 5.2[31] 设 $[a,b]$ 为有限区间或无限区间，如果函数 $\rho(x)$ 满足：

(1) $\rho(x)>0$，$x\subseteq[a,b]$；

(2) $\int_a^b \rho(x)dx>0$；

(3) $\int_a^b x^n\rho(x)dx$ 存在

则称 $\rho(x)$ 为权函数。

定义 5.3[31] 设 $\rho(x)$ 为 $[a,b]$ 上的权函数，对于函数 $f(x)$ 和 $g(x)$，若满足：

$$\int_a^b \rho(x)f(x)g(x)dx=0,$$

则称函数 $f(x)$ 和 $g(x)$ 关于 $\rho(x)$ 正交。

### 5.1.2 函数范数及性质

设已知一列表函数$y_i=f(x_i)(i=0,1,\cdots,m)$。为了构造函数$f(x)$的一个$n(<m)$次近似多项式$p_n(x)$，按最小二乘法，应使和

$$S=\sum_{i=0}^{m}(p_n(x_i)-f(x_i))^2$$

取最小值。这相当于在结点$x_i(i=0,1,\cdots,m)$处约束$p_n(x)$，看$p_n(x)$近似列表函数$f(x)$的程度如何，也只是看在这$m+1$个结点上的情况（亦即平方偏差）

$$((p_n(x_i)-f(x_i))^2).$$

有时也需要考虑在全区间$[a,b]$上构造函数$f(x)$的近似多项式$p_n(x)$，此时自然应以积分

$$\int_a^b(p_n(x)-f(x))^2d(x)$$

代替和$\sum$取最小值。实际上，在数值分析中常以数量

$$\|p-f\|=\sqrt{\int_a^b(p(x)-f(x))^2dx}$$

来度量函数$p(x)$和$f(x)$的接近程度。

只要回想一下$n$维欧氏空间中的两点距离公式，就知道上述数量可以类似的理解为函数空间中的元素$p(x)$和$f(x)$两者间的距离。而当$\|P_n-f\|\to 0(n\to\infty)$时，也就可把$p_n(x)$理解为按照上述的平方度量收敛于$f(x)$，记作

$$P_n(x)\xrightarrow{2}f(x),n\to\infty.$$

在实变函数论中讨论$L^2$空间理论时，人们正是这样来理解一个序列的收敛（或极限）概念的。

为了实用的需要，我们还有必要进一步去扩充上述观点。对于任意一个定义在$[a,b]$上的可测函数$f(x)$，如果$\rho(x)f(x)$为（$L$）可积，则就说$f(x)$属于$L_\rho[a,b]$类；如果$\rho(x)(f(x))^2$为（$L$）可积，则说$f(x)$属于$L_\rho^2[a,b]$类。

现在介绍一下范数的概念。在权函数$\rho(x)=1$下，$L_\rho^2$中的每一个函数$f(x)$，都赋予一个数值

$$\|f\|=\sqrt{\int_a^b[f(x)]^2dx},$$

并称它为$f$的广义绝对值或范数。由此

$$\| f-g \|=\sqrt{\int_a^b [f(x)-g(x)]^2 dx}$$

便给出了两个函数$f$和$g$之间的距离或接近程度的度量。所谓平方逼近就是按照这种度量来衡量其逼近程度的。

函数范数的性质如下：

1. $\| f \| \geq 0$，并且当且仅当$f \equiv 0$时$\| f \|=0$；
2. $\| c \cdot f \|=| c | \cdot \| f \|$，c为任一常数；
3. $\| f+g \| \leq \| f \|+\| g \|$。

## 5.2 函数正交性与函数范数的Mizar实现

从内积的定义与性质入手，在此基础上研究函数正交性的 Mizar 实现，但由于现有 Mizar 文章的局限性，本文仅讨论以$\rho(x)=1$为权函数时函数正交的情况。最后讨论函数范数的 Mizar 描述与证明。

### 5.2.1 函数正交性、函数范数的环境部设置与变量声明

Mizar论文撰写完毕后环境部的最终形式如下：

**environ**

**vocabularies** INTEGRA9, ARYTM, PRE_TOPC, RELAT_1, FUNCT_1, ARYTM_1, JGRAPH_2, PARTFUN1, SEQ_1, SIN_COS, INTEGRA1, FDIFF_1, REALSET1, BOOLE, SQUARE_1, ARYTM_3, ORDINAL2, RCOMP_1, PREPOWER, LATTICES, FINSEQ_1, TAYLOR_1, SIN_COS4, SUBSET_1, GROUP_1, SEQ_4;

**notations** FUNCT_1, SEQ_1, SIN_COS, TARSKI, XBOOLE_0, SUBSET_1, NUMBERS, VALUED_0, VALUED_1, MEMBERED, XREAL_0, SEQ_4, REAL_1, RELSET_1, PARTFUN1, RFUNCT_1, RCOMP_1, PSCOMP_1, INTEGRA1, XXREAL_0, XXREAL_2, FCONT_1, SINCOS10, FDIFF_9, SQUARE_1, INTEGRA5, PARTFUN2, NEWTON, PREPOWER, TAYLOR_1, FDIFF_1, JORDAN16, SEQ_2;

**constructors** SIN_COS, TAYLOR_1, SEQ_1, REAL_1, FDIFF_1, FCONT_1, PSCOMP_1, NAT_1, BINOP_2, SQUARE_1, PREPOWER, INTEGRA5, PARTFUN2, LIMFUNC1, JORDAN16, SEQ_4, RFUNCT_1, FDIFF_9, SINCOS10, POWER;

**registrations** VALUED_1, NAT_1, XBOOLE_0, NUMBERS, MEMBERED, VALUED_0, INTEGRA1, RCOMP_1, INT_1, RELAT_1, ORDINAL1, FUNCT_2, FDIFF_1, COMPLEX1, RFUNCT_1, XREAL_0, RELSET_1, FINSEQ_2, ORDINAL2, NEWTON, SEQM_3, XXREAL_0, FINSEQ_1,

```
    SIN_COS, SQUARE_1, XREAL_1, FUNCOP_1, PARTFUN1, TAYLOR_1,
    JORDAN16, FCONT_3, XXREAL_2;
 requirements NUMERALS, BOOLE, SUBSET, ARITHM;
 definitions SIN_COS, INTEGRA5, VALUED_1, SUBSET_1, TARSKI, FDIFF_9,
    SQUARE_1;
 theorems PARTFUN1, INTEGRA1, RFUNCT_1, FUNCT_1, FDIFF_1, INTEGRA2,
    FUNCT_2, XBOOLE_0, XBOOLE_1, SQUARE_1, XCMPLX_1, INTEGRA5,
    INTEGRA8, INTEGRA6, SIN_COS, XREAL_1, FDIFF_4, TAYLOR_1,
    FDIFF_6, VALUED_1, FCONT_1, JORDAN16, RELAT_1, FDIFF_2, FDIFF_5,
    FDIFF_8, FDIFF_9, PREPOWER;
        ⋮
```

文章中需要声明的变量数据类型如下：

```
begin
 reserve r,x for Real;
 reserve n for Element of NAT;
 reserve A for closed-interval Subset of REAL;
 reserve f,g,f1,f2 for PartFunc of REAL,REAL;
 reserve Z for open Subset of REAL;
```

### 5.2.2 函数正交性的Mizar实现

在讨论函数正交性之前，先给出内积定义的 Mizar 表述：

```
definition
  let A be closed-interval Subset of REAL;
  let f,g be PartFunc of REAL,REAL;
  func |||(f,g,A)||| -> Real equals :: INTEGRA9:def 1
  integral((f(#)g),A);
end;
```

A 为闭区间，f，g 为函数。PartFunc of REAL，REAL 是一种函数的模式，表示函数的定义域（dom）和值域（rng）都包含于实数集（REAL）中。由本文§2.6 中的介绍，建立 INTEGRA9.voc，里面写入“K|||(” 和“L)|||” 以表示新的定义“|||(f,g,A)|||”（参见本文表 2-3）。“INTEGRA9:def 1”是文章 INTEGRA9.miz[28]中的定义 1。

§5.1.1 中所涉及的内积性质在 Mizar 系统中表述为：

性质 1：

```
  theorem :: INTEGRA9:30
    for f,g being PartFunc of REAL,REAL,
        A being closed-interval Subset of REAL holds
```

```
|||(f,g,A)||| = |||(g,f,A)|||;
```

性质 2：

```
theorem :: INTEGRA9:34
for f,g being PartFunc of REAL,REAL,
A being closed-interval Subset of REAL
st (f(#)g)|A is bounded & f(#)g is_integrable_on A
& A c= dom(f(#)g)
 holds |||((r(#)f),g,A)||| = r*|||(f,g,A)|||;
```

其中的 r 为实数，在变量数据类型中已声明。

性质 3：

```
theorem :: INTEGRA9:31
for f1,f2,g being PartFunc of REAL,REAL,
A being closed-interval Subset of REAL st (f1(#)g)||A is total &
(f2(#)g)||A is total & (f1(#)g)||A|A is bounded &
(f1(#)g)||A is integrable & (f2(#)g)||A|A is bounded &
(f2(#)g)||A is integrable
holds |||(f1+f2,g,A)||| = |||(f1,g,A)||| + |||(f2,g,A)|||;
```

性质 4：

```
theorem :: INTEGRA9:32
for f1,f2,g being PartFunc of REAL,REAL,
A being closed-interval Subset of REAL st (f1(#)g)||A is total &
(f2(#)g)||A is total & (f1(#)g)||A|A is bounded &
(f1(#)g)||A is integrable & (f2(#)g)||A|A is bounded &
(f2(#)g)||A is integrable
holds |||(f1-f2,g,A)||| = |||(f1,g,A)||| - |||(f2,g,A)|||;
```

性质 5：

```
theorem :: INTEGRA9:37
for f,g being PartFunc of REAL,REAL,
A being closed-interval Subset of REAL st (f(#)f)||A is total &
(f(#)g)||A is total & (g(#)g)||A is total & (f(#)f)||A|A is bounded &
(f(#)g)||A|A is bounded & (g(#)g)||A|A is bounded
& (f(#)f) is_integrable_on A & (f(#)g) is_integrable_on A
& (g(#)g) is_integrable_on A holds
|||(f+g,f+g,A)||| = |||(f,f,A)||| + 2*|||(f,g,A)||| + |||(g,g,A)|||;
```

以下只给出性质5的Mizar系统证明,其他性质的证明收录于INTEGRA9中。

```
proof
  let f,g be PartFunc of REAL,REAL;
  let A be closed-interval Subset of REAL;
  assume A1: (f(#)f)||A is total
        & (f(#)g)||A is total & (g(#)g)||A is total;
  assume A2: (f(#)f)||A|A is bounded &
     (f(#)g)||A|A is bounded & (g(#)g)||A|A is bounded;
   then A3:((f(#)f)||A+(f(#)g)||A)|(A /\ A) is bounded
       & ((f(#)g)||A+(g(#)g)||A)|(A /\ A) is bounded by RFUNCT_1:100;
  assume (f(#)f) is_integrable_on A
       & (f(#)g) is_integrable_on A & (g(#)g) is_integrable_on A;
  then A4: (f(#)f)||A is integrable & (f(#)g)||A is integrable
     & (g(#)g)||A is integrable by INTEGRA5:def 2;
  then A5: ((f(#)f)||A+(f(#)g)||A) is integrable &
    ((f(#)g)||A+(g(#)g)||A) is integrable by A1,A2,INTEGRA1:59;
  |||(f+g,f+g,A)|||= integral(((f(#)(f+g))+(g(#)(f+g)))||A) by RFUNCT_1:22
 .= integral((((f(#)(f+g))||A)+((g(#)(f+g))||A)) by INTEGRA5:5
 .= integral(((((f(#)f)+(f(#)g))||A)+((g(#)(f+g))||A)) by RFUNCT_1:23
 .= integral(((((f(#)f)+(f(#)g))||A)
          +(((g(#)f)+(g(#)g))||A)) by RFUNCT_1:23
 .= integral((((f(#)f)||A+(f(#)g)||A)
          +(((g(#)f)+(g(#)g))||A)) by INTEGRA5:5
 .= integral((((f(#)f)||A+(f(#)g)||A)
          +((g(#)f)||A+(g(#)g)||A)) by INTEGRA5:5
 .= integral(((f(#)f)||A+(f(#)g)||A)
       +integral((f(#)g)||A+(g(#)g)||A) by A1,A3,A5,INTEGRA1:59
 .= integral((f(#)f)||A)+integral((f(#)g)||A)
       +integral((f(#)g)||A+(g(#)g)||A) by A1,A2,A4,INTEGRA1:59
 .= integral((f(#)f)||A)+integral((f(#)g)||A)
       +(integral((f(#)g)||A)+integral((g(#)g)||A)) by A1,A2,A4,INTEGRA1:59;
  hence thesis;
end;
```

在Mizar论文中，以上内积定义及性质结果表述为[28]:

(30) For all partial functions $f$, $g$ from $\mathbb{R}$ to $\mathbb{R}$ and for every closed-interval subset $A$ of $\mathbb{R}$ holds $|||(f,g,A)||| = |||(g,f,A)|||$.

(34) Let $f$, $g$ be partial functions from $\mathbb{R}$ to $\mathbb{R}$ and $A$ be a closed-interval subset of $\mathbb{R}$. Suppose $(f\,g){\restriction}A$ is bounded and $f\,g$ is integrable on $A$ and $A \subseteq \mathrm{dom}(f\,g)$. Then $|||(r\,f, g, A)||| = r \cdot |||(f, g, A)|||$.

(31) Let $f_1$, $f_2$, $g$ be partial functions from $\mathbb{R}$ to $\mathbb{R}$ and $A$ be a closed-interval subset of $\mathbb{R}$. Suppose that

(i) $(f_1\,g){\restriction}A$ is total,

(ii) $(f_2\,g){\restriction}A$ is total,

(iii) $((f_1\,g){\restriction}A){\restriction}A$ is bounded,

(iv) $f_1\,g$ is integrable on $A$,

(v) $((f_2\,g){\restriction}A){\restriction}A$ is bounded, and

(vi) $f_2\,g$ is integrable on $A$.

Then $|||(f_1 + f_2, g, A)||| = |||(f_1, g, A)||| + |||(f_2, g, A)|||$.

(32) Let $f_1$, $f_2$, $g$ be partial functions from $\mathbb{R}$ to $\mathbb{R}$ and $A$ be a closed-interval subset of $\mathbb{R}$. Suppose that

(i) $(f_1\,g){\restriction}A$ is total,

(ii) $(f_2\,g){\restriction}A$ is total,

(iii) $((f_1\,g){\restriction}A){\restriction}A$ is bounded,

(iv) $f_1\,g$ is integrable on $A$,

(v) $((f_2\,g){\restriction}A){\restriction}A$ is bounded, and

(vi) $f_2\,g$ is integrable on $A$.

Then $|||(f_1 - f_2, g, A)||| = |||(f_1, g, A)||| - |||(f_2, g, A)|||$.

(37) Let $f$, $g$ be partial functions from $\mathbb{R}$ to $\mathbb{R}$ and $A$ be a closed-interval subset of $\mathbb{R}$. Suppose that $(f\,f){\restriction}A$ is total and $(f\,g){\restriction}A$ is total and $(g\,g){\restriction}A$ is total and $((f\,f){\restriction}A){\restriction}A$ is bounded and $((f\,g){\restriction}A){\restriction}A$ is bounded and $((g\,g){\restriction}A){\restriction}A$ is bounded and $f\,f$ is integrable on $A$ and $f\,g$ is integrable on $A$ and $g\,g$ is integrable on $A$. Then $|||(f+g, f+g, A)||| = |||(f, f, A)||| + 2 \cdot |||(f, g, A)||| + |||(g, g, A)|||$.

在 Mizar 中对内积及性质给出了表述，接下来要根据上述内容研究权函数 $\rho(x)=1$ 时的函数正交情况的 Mizar 描述。

Mizar 中函数正交性的定义为：

```
definition
    let A be closed-interval Subset of REAL;
    let f,g be PartFunc of REAL,REAL;
    pred f is_orthogonal_with g,A means :: INTEGRA9:def 2
    |||(f,g,A)||| = 0;
end;
```

这里定义的是谓词(pred)，不需要证明，可直接使用。在 INTEGRA9.voc 中要写入“Ris_orthogonal_with”。此定义为 INTEGRA9.miz 中的定义 2。

由内积性质 5 与函数正交性的定义可得到下面一个定理：

```
theorem :: INTEGRA9:38
 for f,g being PartFunc of REAL,REAL,
 A being closed-interval Subset of REAL st (f(#)f)||A is total &
 (f(#)g)||A is total & (g(#)g)||A is total & (f(#)f)||A|A is bounded &
 (f(#)g)||A|A is bounded & (g(#)g)||A|A is bounded
 & (f(#)f) is_integrable_on A & (f(#)g) is_integrable_on A
 & (g(#)g) is_integrable_on A & f is_orthogonal_with g,A
 holds |||(f+g,f+g,A)||| = |||(f,f,A)||| + |||(g,g,A)|||;
```

Mizar 中证明过程如下：

```
proof
    let f,g be PartFunc of REAL,REAL;
    let A be closed-interval Subset of REAL;
 assume (f(#)f)||A is total & (f(#)g)||A is total
    & (g(#)g)||A is total &(f(#)f)||A|A is bounded &
    (f(#)g)||A|A is bounded & (g(#)g)||A|A is bounded
    & (f(#)f) is_integrable_on A & (f(#)g) is_integrable_on A
    & (g(#)g) is_integrable_on A;
 then A1: |||(f+g,f+g,A)||| = |||(f,f,A)||| + 2*|||(f,g,A)||| + |||(g,g,A)||| by Th37;
 assume f is_orthogonal_with g,A;
 then |||(f,g,A)||| = 0 by Def2;
 hence thesis by A1;
end;
```

其中定理 37（Th37）是内积性质 5，定义 2（Def2）是函数正交性的定义。

以上内容在 Mizar 论文中的表述为[28]：

Let $A$ be a closed-interval subset of $\mathbb{R}$ and let $f$, $g$ be partial functions from $\mathbb{R}$ to $\mathbb{R}$. We say that $f$ is orthogonal with $g$, $A$ if and only if:

(Def. 2) $|||(f, g, A)||| = 0$.

(38) Let $f$, $g$ be partial functions from $\mathbb{R}$ to $\mathbb{R}$ and $A$ be a closed-interval subset of $\mathbb{R}$. Suppose that $(f\,f) \restriction A$ is total and $(f\,g) \restriction A$ is total and $(g\,g) \restriction A$ is total and $((f\,f) \restriction A) \restriction A$ is bounded and $((f\,g) \restriction A) \restriction A$ is bounded and $((g\,g) \restriction A) \restriction A$ is bounded and $f\,f$ is integrable on $A$ and $f\,g$ is integrable on $A$ and $g\,g$ is integrable on $A$ and $f$ is orthogonal with $g$, $A$. Then $|||(f+g, f+g, A)||| = |||(f, f, A)||| + |||(g, g, A)|||$.

讨论函数正交性时需要给出区间，在特定的区间上讨论函数的正交情况。下面给出某些函数的正交性在 Mizar 中的推证。

函数 $y=\sin x$ 和 $y=\cos x$ 在区间 $[0,\pi]$ 上正交，在 Mizar 中此定理表述为：

theorem
A = [.0,PI.] implies sin is_orthogonal_with cos,A

证明如下：

```
proof
  assume A1: A = [.0,PI.];
  |||(sin,cos,A)||| = 0 by INTEGRA8:92,A1;
  hence thesis by Def2;
end;
```

同理，函数 $y=\sin x$ 和 $y=\cos x$ 在区间 $[0,2\pi]$，$[2n\pi,(2n+1)\pi]$，$[x+2n\pi,x+(2n+1)\pi]$（$x$ 为任意实数），$\left[-\frac{\pi}{2},\frac{\pi}{2}\right]$，$[-\pi,\pi]$，$[-2\pi,2\pi]$，$[-2n\pi,2n\pi]$，$[x-2n\pi,x+2n\pi]$（$x$ 为任意实数）上也正交，Mizar 中表述分别为：

theorem
A = [.0,PI*2.] implies sin is_orthogonal_with cos,A
theorem
A = [.2*n*PI,(2*n+1)*PI.] implies sin is_orthogonal_with cos,A
theorem
A = [.x+2*n*PI,x+(2*n+1)*PI.] implies sin is_orthogonal_with cos,A
theorem
A = [.-PI/2,PI/2.] implies sin is_orthogonal_with cos,A
theorem
A = [.-PI,PI.] implies sin is_orthogonal_with cos,A
theorem
A = [.-2*PI,2*PI.] implies sin is_orthogonal_with cos,A
theorem
A = [.-2*n*PI,2*n*PI.] implies sin is_orthogonal_with cos,A
theorem
A = [.x-2*n*PI,x+2*n*PI.] implies sin is_orthogonal_with cos,A

以上定理的证明过程与函数 $y=\sin x$ 和 $y=\cos x$ 在区间 $[0,\pi]$ 上正交类似；参 INTEGRA9.miz[28]。

以上所讨论的函数的正交性在 Mizar 论文中的表述如下[28]：

(40) If $A = [0, \pi]$, then the function sin is orthogonal with the function cos, $A$.

(41) If $A = [0, \pi \cdot 2]$, then the function sin is orthogonal with the function cos, $A$.

(42) If $A = [2 \cdot n \cdot \pi, (2 \cdot n + 1) \cdot \pi]$, then the function sin is orthogonal with the function cos, $A$.

(43) If $A = [x + 2 \cdot n \cdot \pi, x + (2 \cdot n + 1) \cdot \pi]$, then the function sin is orthogonal with the function cos, $A$.

(45) If $A = [-\frac{\pi}{2}, \frac{\pi}{2}]$, then the function sin is orthogonal with the function cos, $A$.

(44) If $A = [-\pi, \pi]$, then the function sin is orthogonal with the function cos, $A$.

(46) If $A = [-2 \cdot \pi, 2 \cdot \pi]$, then the function sin is orthogonal with the function cos, $A$.

(47) If $A = [-2 \cdot n \cdot \pi, 2 \cdot n \cdot \pi]$, then the function sin is orthogonal with the function cos, $A$.

(48) If $A = [x - 2 \cdot n \cdot \pi, x + 2 \cdot n \cdot \pi]$, then the function sin is orthogonal with the function cos, $A$.

### 5.2.3 函数范数的Mizar实现

Mizar 中函数范数定义表述为：

```
definition
  let A be closed-interval Subset of REAL;
  let f be PartFunc of REAL,REAL;
  func ||..f,A..|| -> Real equals :: INTEGRA9:def 3
    sqrt (|||(f,f,A)|||);
end;
```

在 INTEGRA9.voc 中写入“K||..”和“L..||”以表示新的定义“||..  ..||”。Sqrt 表示开方。

本文仅讨论 § 5.1.2 中函数范数性质 1 及 $c=1$ 时性质 2 的 Mizar 表述与证明。

性质 1：

```
theorem
  for f being PartFunc of REAL,REAL,
  A being closed-interval Subset of REAL st
  (f(#)f)||A is total & (f(#)f)||A|A is bounded &
  (f(#)f)||A is integrable &
  (for x st x in A holds (f(#)f)||A.x >= 0) holds 0 <= ||..f,A..||
```

证明过程如下：

```
  proof
    let f be PartFunc of REAL,REAL;
    let A be closed-interval Subset of REAL;
    assume A1: (f(#)f)||A is total;
    assume A2:(f(#)f)||A|A is bounded & (f(#)f)||A is integrable
```

```
      & (for x st x in A holds (f(#)f)||A.x >= 0);
    |||(f,f,A)||| >= 0 by A1,A2,INTEGRA2:32;
    hence thesis by SQUARE_1:def 4;
end;
```

性质 2：

```
theorem
  for f being PartFunc of REAL,REAL,
  A being closed-interval Subset of REAL holds
  ||..(1(#)f),A..|| = ||..f,A..|| by RFUNCT_1:33;
```

上述定理的划线部分表明 Mizar 系统可以根据 Mizar 数据库里的数学知识自动证明。

上述内容在 Mizar 论文中的表述为[28]：

(49) Let $f$ be a partial function from $\mathbb{R}$ to $\mathbb{R}$ and $A$ be a closed-interval subset of $\mathbb{R}$. Suppose $(f\,f)\restriction A$ is total and $((f\,f)\restriction A)\restriction A$ is bounded and $f\,f$ is integrable on $A$ and for every $x$ such that $x \in A$ holds $((f\,f)\restriction A)(x) \geq 0$. Then $0 \leq ||..f.\,A..||$.

(50) For every partial function $f$ from $\mathbb{R}$ to $\mathbb{R}$ and for every closed-interval subset $A$ of $\mathbb{R}$ holds $||..1\,f.\,A..|| = ||..f.\,A..||$.

下面的定理在 Mizar 中由函数的正交性和函数范数共同得到。

定理　当函数 $f(x)$ 和 $g(x)$ 在给定区间 $A$ 上正交时，

$$\int_A (f+g)^2 dx = \int_A f^2 dx + \int_A g^2 dx$$

在 Mizar 中的表述为：

```
theorem
  for f,g being PartFunc of REAL,REAL,
  A being closed-interval Subset of REAL st (f(#)f)||A is total &
  (f(#)g)||A is total & (g(#)g)||A is total & (f(#)f)||A|A is bounded
  & (f(#)g)||A|A is bounded & (g(#)g)||A|A is bounded
  & (f(#)f) is_integrable_on A & (f(#)g) is_integrable_on A
  & (g(#)g) is_integrable_on A & f is_orthogonal_with g,A
  & (for x st x in A holds (f(#)f)||A.x >= 0)
  & (for x st x in A holds (g(#)g)||A.x >= 0)
  holds ||..(f+g),A..||^2 = ||..f,A..||^2 + ||..g,A..||^2
```

在 Mizar 系统中完整的证明过程如下：

```
proof
```

```
    let f,g be PartFunc of REAL,REAL;
    let A be closed-interval Subset of REAL;
  assume A1: (f(#)f)||A is total & (f(#)g)||A is total
    & (g(#)g)||A is total & (f(#)f)||A|A is bounded &
    (f(#)g)||A|A is bounded & (g(#)g)||A|A is bounded
    & (f(#)f) is_integrable_on A & (f(#)g) is_integrable_on A
    & (g(#)g) is_integrable_on A;
  then B1: (f(#)f)||A is integrable & (f(#)g)||A is integrable
    & (g(#)g)||A is integrable by INTEGRA5:def 2;
  assume A2: f is_orthogonal_with g,A;
  A3: |||(f+g,f+g,A)||| = |||(f,f,A)||| + |||(g,g,A)||| by A1,A2,Th38;
  assume (for x st x in A holds (f(#)f)||A.x >= 0);then
  A4: |||(f,f,A)||| >= 0 by A1,B1,INTEGRA2:32;
  assume (for x st x in A holds (g(#)g)||A.x >= 0);then
  A5: |||(g,g,A)||| >= 0 by A1,B1,INTEGRA2:32;
    |||((f+g),(f+g),A)||| >=0 by A3,A4,A5,XREAL_1:35; then
  A6: ||..(f+g),A..||^2 = |||((f+g),(f+g),A)||| by SQUARE_1:def 4;
  A7: ||..f,A..||^2 = |||(f,f,A)||| by A4,SQUARE_1:def 4;
      ||..g,A..||^2 = |||(g,g,A)||| by A5,SQUARE_1:def 4;
  hence thesis by A1,A2,Th38,A6,A7;
end;
```

其中定义 2（Def2）是函数正交性的定义，定理 38（Th38）是函数正交性的性质 5。

此定理的 Mizar 描述为[28]：

(51) Let $f$, $g$ be partial functions from $\mathbb{R}$ to $\mathbb{R}$ and $A$ be a closed-interval subset of $\mathbb{R}$. Suppose that $(f\,f) \upharpoonright A$ is total and $(f\,g) \upharpoonright A$ is total and $(g\,g) \upharpoonright A$ is total and $((f\,f) \upharpoonright A) \upharpoonright A$ is bounded and $((f\,g) \upharpoonright A) \upharpoonright A$ is bounded and $((g\,g) \upharpoonright A) \upharpoonright A$ is bounded and $f\,f$ is integrable on $A$ and $f\,g$ is integrable on $A$ and $g\,g$ is integrable on $A$ and $f$ is orthogonal with $g$, $A$ and for every $x$ such that $x \in A$ holds $((f\,f) \upharpoonright A)(x) \geq 0$ and for every $x$ such that $x \in A$ holds $((g\,g) \upharpoonright A)(x) \geq 0$. Then $||..f+g, A..||^2 = ||..f, A..||^2 + ||..g, A..||^2$.

## 5.3 Mizar实现的最终形式

关于函数正交性与函数范数的论文INTEGRA9.miz，若经系统中的所有校验和优化命令的检验均无错误，即为Mizar实现的最终形式。INTEGRA9.miz经所有系统命令检测后的最终结果见图5-1：




图5-1：INTEGRA9.miz的mizf检验结果

Fig.5-1 the result of INTEGRA9.miz

文章INTEGRA9.miz经过mizf命令的检验，在DOS下显示结果为：环境部包含8部分，全文共2426行，检验总耗时37秒，Parser、Analyzer、Checker三个过程均正确无误。

另外，函数正交性与函数范数Mizar实现的文本形式[28]收录于2009年的***Formalized Mathematics***。

## 5.4 成果出处

本文关于函数正交性与函数范数的相关结果收录在 Mizar 版本 Mizar version: 7.10.01_4.112.1041 中。

# 总结与展望

科学技术的现代化，离不开数学的现代化；而数学的现代化，在某种意义上也可以说是数学的机械化。定理机器证明是其主要内容之一，也是涉及人类智能问题的主要研究课题之一。从传统的手工证明到定理机器证明是现代数学思想方法的一次重大突破。

在定理机器证明领域，计算机的识别和推理功能尤为重要。Mizar系统在计算机上将这两个功能很好的结合在了一起。人们用Mizar系统中规定的描述方法将定理的证明过程表达出来，就可对其进行验证。如果在证明过程中出现了数据库（MML）中没有的定理，系统就会给出提示并进行验证。若在DOS下验证发现证明过程存在错误，系统会立即由数字标明出错的位置，并在定理末尾用特定的数字表示逻辑中错误的原因。经Mizar系统中mizf命令和所有优化命令验证后没有错误，便可申请加入Mizar数据库。加入数据库的定理需要经过数据库管理者们的评估，确定后才能给定理命名。除此之外，数据库中还可保存一些有待证明的重要猜想。无论是谁，只要给出了猜想的正确证明，系统就会马上宣布该难题被攻破。

经过三十多年的研究与发展，Mizar系统建立了完整的知识结构和坚实的理论基础，创造了既简单易懂又语法严谨的Mizar数学语言，形成了庞大的数学知识数据库，完成并不断完善Mizar数学大百科全书。同其他数学机器证明软件相比，Mizar系统能够更直接、更快速地发现证明推理过程中出现的矛盾性、不明确性、不完整性、不一致性及不相容性。目前最新版本的Mizar系统可以识别出近八百种性质不同的错误，从而保证了推理论证的可靠性、逻辑性、严密性和科学性。随着Mizar数据库（MML）的不断扩充，Mizar仍在向前发展。

本文基于对Mizar语言的学习和Mizar系统的研究，讨论了如何在Mizar中实现特殊函数差分差商、特殊函数积分、函数正交性与函数范数及相关内容，借助Mizar的数学知识体系及理论基础，实现了它们在Mizar系统中的逻辑推理论证，为Mizar系统在分析学、数值计算等数学领域的应用提供了理论依据和实践基础。

目前Mizar系统的研究工作有以下几个方面：

一、充实Mizar数据库（MML）。几乎所有数学问题都可在Mizar系统下进行证明、推理和计算，所以发展现有理论的同时，还需要不断充实Mizar数据库中的理论知识。

二、完善Mizar语言，研发新的证明方法。随着Mizar数据库内容的不断扩充，Mizar语言越来越趋于简捷化、人性化。同时为于简化证明过程、提高检验效率，

需要研发适应Mizar系统的新的证明方法。

三、改进Mizar系统。一方面，在定理证明过程中，应允许“同理可得”，以避免造成整个定理的冗长。另一方面，应把某些基础的定义、定理与性质设定为系统默认，从而省略推理依据的引用，提高系统效率。

四、完善Mizar教学功能。Mizar作为一种新兴的数学教学工具，早在1985年华沙大学的几何课中就实现了其教学功能[19]。随着Mizar系统的发展，其逻辑推理功能在数学各分支的教学中都有体现，特别是在波兰、加拿大等西方国家的数学教学中。

五、加强与其它学科、软件的结合。目前，Mizar在自动化控制、音像识别系统也有应用，与计算机科学相结合，参与人工智能和图像处理的实现过程。

Mizar基于纯数学的理论体系已经形成，对于数学各个分支的细节部分在Mizar系统中的实现与发展仍有待探讨。此外，与其它学科交叉结合也有待深入，并努力实现向OpenMath（OMDoc）等方向的转换。

数学机械化程序的提高，将促进科学技术现代化的程度，并由此促进生产力的提高，从而使我国的社会主义建设获得飞跃的发展。

# 参考文献


[1] 张红，数学简史，北京：科学出版社，2007．1

[2] 高小山，数学机械化研究展望，中国基础科学，1999：26-28

[3] 殷堰工，数学定理的机器证明，科技潮，1997，11期：11

[4] 骆祖英，定理机器证明与传统数学观，浙江师大学报，1996，19(4)：1-4

[5] 田长生，几何定理的机器证明——每个中国数学教师都应懂得的方法，广东技术师范学院学报，2003，6：82-84

[6] <http://www.cs.ru.nl/~freek/digimath/>

[7] R.P. Nederpelt, J.H. Geuvers, R.C. de Vrijer, Selected Papers on Automath (Studies in Logic and the Foundations of Mathematics)[M], US:Elsevier Publishing Company, 1994

[8] M. Muzalewski, An Outline of PC Mizar[M], Brussels: Fondation Philippe le Hodey, 1993,<http://www.cs.kun.nl/~freek/mizar/mizarmanual.ps.gz>

[9] F. Wiedijk, Mizar: An Impression[M], 1999, <http://www.cs.ru.nl/~freek/mizar/mizarintro.pdf>

[10] Robert L. Constable, Stuart F. Allen, H.M. Bromley, et.al. Implementing Mathematics with the Nuprl Development System., Prentice-Hall, NJ, 1986

[11] B. Buchberger, T. Jebelean, F. Kriftner, et.al, An Overview on the Theorema project, In W. Kuechlin, editor, Proceedings of ISSAC'97 (International Symposium on Symbolic and Algebraic Computation), Maui, Hawaii,1997, ACM Press

[12] Matt Kaufmann, Panagiotis Manolios and J. Strother Moore, Computer-Aided Reasoning: An Approach, Kluwer Academic Publishers, Boston, 2000

[13] S. Owre, J. Rushby and N. Shankar, PVS: A prototype verification system, In D. Kapur, editor, 11th International Conference on Automated Deduction(CADE), volume 607 of LNAI, Berlin, Heidelberg, New York, 1992, SpringerVerlag, 748-752

[14] Massimo Marchiori, The Mathematical Semantic Web, A.Asperti, B.Buchberger, J.K.Davenport(Eds): MKM2003, LNCS2594, 2003, 16-223

[15] F. Wiedijk, The Seventeen Provers of the World, Springer Berlin / Heidelberg, 2006

[16] Yatsuka Nakamura. Mizar Lecture Notes. Publisher: Shinshu University, Department of Information Engineering, Kiso Lab,2002

[17] Piotr Rudnichi, An Overview of the MIZAR Project, Chalmers University of Technology, Bastad, 1992

[18] 丁玉忠，Mizar 语言的开发与应用，[学位论文]：青岛科技大学，2005

[19] R. Matuszewski and P. Rudnicki, Mizar: the first 30 years, 2004 <http:/merak.pb.bialystok.pl/mkm2004/>

[20] http://www.mizar.org Mizar官方网站

[21] <http://markun.cs.shinshu-u.ac.jp/kiso/projects/proofchecker/mizar/mizardictionary1.htm>

[22] Yatsuka Nalcamura, Mizar Lecture Notes, Publisher: Shinshu University, Department of Information Engineering, Kiso Lab, 4th Edition Published in Oct 1st, 2002

[23] 王仁宏，数值逼近，北京：高等教育出版社，2003．93-98

[24] Bo Li, Yanping Zhuang, Xiquan Liang, Difference and Difference Quotient--Part II, Formalized Mathematics, 2008, 16 (1):45-50

[25] 华东师范大学数学系，数学分析，北京：高等教育出版社，2003．177-204

[26] 复旦大学数学系，数学分析（上册），北京：高等教育出版社，1983.

[27] Dr. Fritz Chemnitius, Differentiation und Integration Ausgewählter Beispiele, Zweite Auflage, Veb Verlag Technik, Berlin, 1956

[28] Bo Li, Yanping Zhuang, Bing Xie, Several Integrability Formulas of Some Functions, Orthogonal Polynomials and Norm Functions, Formalized Mathematics, 2009,17(1):11-21

[29] Bo Li, Yanping Zhuang, Yanhong, Men, Several Integrability Formulas of Special Functions --Part {II}, Formalized Mathematics, 2009,17(1):23-35

[30] 陈传璋，金福临等，数学分析，北京：高等教育出版社，1978.

[31] 李文林，数学史概论，北京：高等教育出版社，2002．330-338

[32] 钱宝宗，中国数学史，北京：科学出版社，1964.

[33] 林艺，数学小百科，北京：机械工业出版社，1999．256-257

[34] 北京矿业学院高等教育教研组，数学手册北京：燃料化学工业出版社，1973.

[35] 李岳生，黄友谦，数值逼近，北京：人民教育出版社，1978.

[36] 吴文俊，数学机械化，北京：科学出版社，2003.

[37] 吴文俊，几何定理机器证明的基本原理，北京：科学出版社，1984.

[38] Czeslaw Bylinski, Functions and Their Basic Properties, Formalized Mathematics, 1990,1(1):55-65

[39] Czeslaw Bylinski, Functions from a Set to a Set, Formalized Mathematics, 1990,1(1):153-164

[40] Czeslaw Bylinski, Partial Functions, Formalized Mathematics, 1990,1(2):357-367

[41] Andrzej Trybulec and Czeslaw Bylinski, Some Properties of Real Numbers, Formalized Mathematics, 1990,1(3):445-449

[42] Konrad Raczkowski and Pawe Sadowski, Real Function Differentiability, Formalized Mathematics, 1990,1(4):797-801

[43] Jaroslaw Kotowicz, Partial Functions from a Domain to the Set of Real Numbers, Formalized Mathematics, 1990,1(4):703-709

[44] Konrad Raczkowski, Pawel Sadowski, Real Function Continuity, Formalized Mathematics, 1990,1(4):787-791

[45] Konrad Raczkowski, Integer and Rational Exponents, Formalized Mathematics 1991,2(1):125-130

[46] Jaroslaw Kotowicz, Konrad Raczkowski, Real Function Differentiability -- Part II, Formalized Mathematics, 1991,2(3): 407-411

[47] Yan Zhang, Xiquan Liang, Several Differentiable Formulas of Special Functions, Formalized Mathematics, 2005,13(3):427-434

[48] Jianbing Cao, Fahui Zhai, Xiquan Liang, Some Differentiable Formulas of Special Functions, Formalized Mathematics, 2005,13(4):505-509

[49] Bo Li, Yan Zhang, Xiquan Liang, Several Differentiation Formulas of Special Functions. Part III, Formalized Mathematics, 2006,14(1):37-45

[50] Bo Li, Peng Wang, Several Differentiation Formulas of Special Functions. Part IV, Formalized Mathematics, 2006,14(3):109-114

[51] Peng Wang, Bo Li, Several Differentiation Formulas of Special Functions. Part {V}, Formalized Mathematics, 2007,15(3):73-79

[52] Bo Li, Pan Wang, Several Differentiation Formulas of Special Functions. Part {VI},

Formalized Mathematics, 2007,15(4):243-250

[53] Yuguang Yang, Yasunari Shidama, Trigonometric Functions and Existence of Circle Ratio, Formalized Mathematics, 1998,7(2):255-263

[54] Takashi Mitsuishi, Yuguang Yang, Properties of the Trigonometric Function, Formalized Mathematics, 1999,8(1):103-106

[55] Takashi Mitsuishi, Noboru Endou, Keiji Ohkubo, Trigonometric Functions on Complex Space, Formalized Mathematics, 2003,11(1):29-32

[56] Pacharapokin Chanapat, Kanchun,, Hiroshi Yamazaki, Formulas and Identities of Trigonometric Functions, Formalized Mathematics, 2004,12(2):139-141

[57] Yuzhong Ding, Xiquan Liang, Formulas and Identities of Trigonometric Functions, Formalized Mathematics, 2004,12(3):243-246

[58] Artur Kornilowicz, Yasunari Shidama.,Inverse Trigonometric Functions Arcsin and Arccos, Formalized Mathematics, 2005,13(1):73-79

[59] Fuguo Ge, Xiquan Liang, Yuzhong Ding, Formulas and Identities of Inverse Hyperbolic Functions, Formalized Mathematics, 2005,13(3):383-387

[60] Pacharapokin Chanapat, Hiroshi Yamazaki, Formulas and Identities of Hyperbolic Functions, Formalized Mathematics, 2005,13(4):511-513

[61] Xiquan Liang, Bing Xie, Inverse Trigonometric Functions Arctan and Arccot, Formalized Mathematics, 2008,16(2):149-159

[62] Bing Xie, Xiquan Liang and Fuguo Ge, Inverse Trigonometric Functions Arcsec1, Arcsec2, Arccosec1 and Arccosec2, Formalized Mathematics, 2008,16(2):161-167

[63] Noboru Endou, Artur Kornilowicz, The Definition of the Riemann Definite Integral and some Related Lemmas, Formalized Mathematics, 1999, 8(1):93-102

[64] Noboru Endou, Katsumi Wasaki, Yasunari Shidama. Scalar Multiple of Riemann Definite Integral, Formalized Mathematics, 2001,9(1):191-196

[65] Noboru Endou, Katsumi Wasaki, Yasunari Shidama, Darboux's Theorem, Formalized Mathematics, 2001,9(1):197-200

[66] Noboru Endou, Katsumi Wasaki, Yasunari Shidama, Integrability of Bounded Total Functions, Formalized Mathematics, 2001,9(2):271-274

[67] Noboru Endou, Katsumi Wasaki, Yasunari Shidama, Definition of Integrability for Partial Functions from REAL to REAL and Integrability for Continuous Functions, Formalized Mathematics, 2001,9(2):281-284

[68] Noboru Endou, Yasunari Shidama, Masahiko Yamazaki, Integrability and the Integral of Partial Functions from $\Bbb R$ into $\Bbb R$, Formalized Mathematics, 2006,14(4):207-212

[69] Yasunari Shidama, Noboru Endou, Katsumi Wasaki, Riemann Indefinite Integral of Functions of Real Variable, Formalized Mathematics, 2007,15(2):59-63

[70] Cuiying Peng, Fuguo Ge, Xiquan Liang, Several Integrability Formulas of Special Functions, Formalized Mathematics, 2007,15(4):189-198

[71] Czeslaw Bylinski, Some Basic Properties of Sets, Formalized Mathematics,1990,1(1):47-53

[72] Andrzej Trybulec, Yatsuka Nakamura, On the Decomposition of a Simple Closed Curve into Two Arcs, Formalized Mathematics, 2002,10(3):163-167

[73] Xiquan Liang, Solving Roots of Polynomial Equations of Degree 2 and 3 with Real Coefficients, Formalized Mathematics, 2001,9(2):347-350

[74] Yasunari Shidama, The Taylor Expansions, Formalized Mathematics, 2004,12(2):195-200
[75] Bo Li, Yan Zhang, Xiquan Liang, Difference and Difference Quotient, Formalized Mathematics, 2006,14(3):115-119

# 致 谢

衷心感谢导师李博教授在我研究生学习的三年时间里对我的关怀和帮助。他在本文选题、内容研究和文章撰写过程中都给予我细心的指导，并提出了许多宝贵的意见。他渊博的学识，严谨的治学作风和一丝不苟的教学精神，使我受益匪浅。

感谢梁希泉教授、波兰华沙大学的Andrzej Trybulec教授和Adam Grabowski教授以及日本信州大学的Yatsuka Nakamura教授在我学习和研究工作中的帮助和支持。

同时，感谢所有在我的研究工作中给予帮助的老师和同学，正是在他们的帮助下，我才得以顺利地完成研究工作。

# 攻读学位期间发表的学术论文目录

已发表的论文：

[1] Bo Li, Yanping Zhuang, Xiquan Liang, Difference and Difference Quotient--Part II, Formalized Mathematics (Poland), 2008, 16 (1):45-49

[2] Bo Li, Yanping Zhuang, Bing Xie, Pan Wang, Several Integrability Formulas of Some Functions, Orthogonal Polynomials and Norm Functions, Formalized Mathematics (Poland), 2009, 17 (1):11-21

[3] Bo Li, Yanping Zhuang, Yanhong Men, Several Integrability Formulas of Special Functions --Part {II}, Formalized Mathematics (Poland), 2009, 17 (1):23-35

[4] Bo Li, Pan Wang, Xiquan Liang, Yanping Zhuang, Some Operations on Quaternion Numbers, Formalized Mathematics (Poland), 2009,17(2):61-65

审稿中的论文：

[5] Bo Li, Yanping Zhuang, Several Integrability Formulas of Special Functions -- Part {III}